

FPR

Functional Programming

Introduction to Functional Programming

Jeremy Gibbons

University of Oxford

February 2015

Part 0

Course aims and objectives

0.0 Outline

Aims

Motivation

Contents

What's it all about?

Literature

0.1 Aims

- *functional programming is programming with values: value-oriented programming*
- no ‘actions’, no side-effects — a radical departure from ordinary (imperative or OO) programming
- surprisingly, it is a powerful (and fun!) paradigm
- better ways of gluing programs together: *component-oriented programming*
- ideas are applicable in ordinary programming languages too; aim to introduce you to the ideas, to improve your day-to-day programming
- (I don’t expect you all to start using functional languages)

0.2 Motivation

LISP is worth learning [because of] the profound enlightenment experience you will have when you finally get it. That experience will make you a better programmer for the rest of your days, even if you never actually use LISP itself a lot.

Eric S. Raymond, American computer programmer (1957–)
How to Become a Hacker

www.catb.org/~esr/faqs/hacker-howto.html

You can never understand one language until you understand at least two.

Ronald Searle, British artist (1920–2011)

0.3 Contents

1. Programming with expressions and values
2. Types and polymorphism
3. Lists
4. Algebraic datatypes
5. Higher-order programming
6. Laziness
7. Reasoning and calculating
8. Monads

0.4 Expressions vs statements

- in ordinary programming languages the world is divided into a world of statements and a world of expressions

- statements:

- ▶ $x := E$, $s1 ; s2$, **while** b **do** s
- ▶ evaluation order is important

$i := i + 1 ; a := a * i \neq a := a * i ; i := i + 1$

- expressions:

- ▶ eg $a + b * c$, a and not b
- ▶ evaluation order is unimportant (assuming no side-effects):
in $(2 * a * y + b) * (2 * a * y + c)$, evaluate either parenthesis first (or both together!)

0.4 Optimizations

- useful optimizations:

- ▶ branch merging:

```
    if b then p else p end  
=    p
```

- ▶ common subexpression elimination:

```
    z := (2*a*y+b)*(2*a*y+c)  
=    t := 2*a*y ; z := (t+b)*(t+c)
```

- ▶ parallel execution: evaluate subexpressions concurrently

- most optimizations require *referential transparency*

- ▶ all that matters about the expression is its value
- ▶ follows from ‘no side effects’
- ▶ ... which follows from ‘no :=’
- ▶ with assignments, side-effect-freeness is very hard to check

0.4 Programming with expressions

- expressions are much shorter and simpler than the corresponding statements
- eg compare using expression:

```
z := (2*a*y+b)*(2*a*y+c)
```

with not using expressions:

```
ac := 2; ac *= a; ac *= y; ac += b; t := ac;  
ac := 2; ac *= a; ac *= y; ac += c; ac *= t;  
z := ac
```

- but in order to discard statements, the expression language must be extended
- functional programming is *programming with an extended expression language*

0.4 Comparison with ‘ordinary’ programming

- insertion sort
- quicksort

0.4 Insertion sort

```
insertSort [ ]      = [ ]  
insertSort (x: xs) = insert x (insertSort xs)  
insert a [ ]        = [ a ]  
insert a (b: xs)  
  | a ≤ b           = a:b:xs  
  | otherwise       = b:insert a xs
```

```
PROCEDURE InsertSort(VAR a:ArrayT);
VAR i, j: CARDINAL;
    t: ElementT;
BEGIN
  FOR i := 2 TO Size DO
    (* a[1..i-1] already sorted *)
    t := a[i];
    j := i;
    WHILE (j > 1) AND (a[j-1] > t) DO
      a[j] := a[j-1]; j := j-1
    END;
    a[j] := t
  END
END InsertSort;
```

0.4 Quicksort

```
quickSort [ ]      = [ ]  
quickSort (x : xs) = quickSort littles ++ [ x ] ++ quickSort bigs  
  where littles    = [ a | a ← xs, a < x ]  
        bigs       = [ a | a ← xs, a ≥ x ]
```

```
void quicksort(int a[], int l, int r)
{
    if (r > l)
    {
        int i = l; int j = r;
        int p = a[(l + r) / 2];
        for (;;) {
            while (a[i] < p) i++;
            while (a[j] > p) j--;
            if (i > j) break;
            swap(&a[i++], &a[j--]);
        };
        quicksort(a, l, j);
        quicksort(a, i, r);
    }
}
```

0.5 Literature

- Richard Bird, *Thinking Functionally with Haskell*, Cambridge University Press, 2015.
- Miran Lipovaca, *Learn You a Haskell for Great Good!: A Beginner's Guide*, No Starch Press, 2011.
- Paul Hudak, *The Haskell School of Expression: Learning Functional Programming through Multimedia*, Cambridge University Press, 2000.
- Graham Hutton, *Programming in Haskell*, Cambridge University Press, 2007.
- Bryan O'Sullivan, John Goerzen, Don Stewart, *Real World Haskell*, O'Reilly Media, 2008.
- Simon Thompson, *Haskell: The Craft of Functional Programming (3rd Edition)*, Addison-Wesley Professional, 2011.

Part 1

Programming with expressions and values

1.0 Outline

Scripts and sessions

Evaluation

Functions

Definitions

Summary

1.1 Calculators

- functional programming is like using a pocket calculator
- user enters in expression, the system evaluates and prints result
- interactive ‘read-eval-print’ loop
- powerful mechanism for defining new functions
- we can calculate not only with numbers, but also with lists, trees, pictures, music ...

1.1 Scripts and sessions

- we will use *GHCI*, an interactive version of the *Glasgow Haskell Compiler*, a popular implementation of the standard lazy functional programming language *Haskell*
- program is a collection of modules
- a module is a collection of definitions: a *script*
- running a program consists of loading script and evaluating expressions: a *session*
- a standalone program includes a ‘main’ expression
- scripts may or may not be *literate* (emphasis on comments)

1.1 An illiterate script

```
-- compute the square of an integer
square :: Integer -> Integer
square x = x * x

-- smaller of two arguments
smaller :: (Integer, Integer) -> Integer
smaller (x, y) = if x <= y then x else y
```

1.1 A literate script

The following function squares an integer.

```
> square :: Integer -> Integer
> square x = x * x
```

This one takes a pair of integers as an argument, and returns the smaller of the two as a result. For example,

```
smaller (3, 4) = 3
```

```
> smaller :: (Integer, Integer) -> Integer
> smaller (x, y) = if x <= y then x else y
```

1.1 Layout

- elegant and unobtrusive syntax
- structure obtained by layout, not punctuation
- all definitions in same scope must start in the same column
- indentation from start of definition implies continuation

```
smaller:: (Integer, Integer) → Integer  
smaller (x, y)  
  = if  
    x ≤ y  
  then  
    x  
  else  
    y
```

- blank lines around code in literate script!
- use spaces, not tabs!

1.1 A session

? 42

42

? 6 * 7

42

? *square* 7 - *smaller* (3, 4) - *square* (*smaller* (2, 3))

42

? *square* 1234567890

1524157875019052100

1.2 Evaluation

- interpreter evaluates expression by reducing to simplest possible form
- reduction is rewriting using meaning-preserving simplifications: replacing equals by equals

$\text{square } (3 + 4)$
 \Rightarrow { definition of $+$ }
 $\text{square } 7$
 \Rightarrow { definition of square }
 $7 * 7$
 \Rightarrow { definition of $*$ }
 49

- expression 49 cannot be reduced any further: *normal form*
- *applicative order* evaluation: reduce arguments before expanding function definition (call by value, eager evaluation)

1.2 Alternative evaluation orders

- other evaluation orders are possible:

$square\ (3 + 4)$
 \Rightarrow { definition of $square$ }
 $(3 + 4) * (3 + 4)$
 \Rightarrow { definition of $+$ }
 $7 * (3 + 4)$
 \Rightarrow { definition of $+$ }
 $7 * 7$
 \Rightarrow { definition of $*$ }
 49

- final result is the same: if two evaluation orders terminate, both yield the same result (*confluence*)
- *normal order* evaluation: expand function definition before reducing arguments (call by need, lazy evaluation)

1.2 Non-terminating evaluations

- consider script

three :: *Integer* → *Integer*
three _ = 3

infinity :: *Integer*
infinity = 1 + *infinity*

- two different evaluation orders:

<i>three infinity</i>		
⇒ { definition of <i>infinity</i> }		
<i>three</i> (1 + <i>infinity</i>)		<i>three infinity</i>
⇒ { definition of <i>infinity</i> }	⇒	{ definition of <i>three</i> }
<i>three</i> (1 + (1 + <i>infinity</i>))		3
⇒ ...		

- not all evaluation orders terminate, even on the same expression; Haskell uses lazy evaluation

1.2 Values

- in FP, as in maths, the sole purpose of an expression is to denote a value
- other characteristics (time to evaluate, number of characters, etc) are irrelevant
- values may be of various kinds: numbers, truth values, characters, tuples, lists, functions, etc
- important to distinguish *abstract value* (the number 42) from concrete representation (the characters '4' and '2', the string "XLII", the bitsequence 0000000000101010)
- evaluator prints *canonical representation* of value
- some values have no canonical representation (eg functions), some have only infinite ones (eg π)

1.2 Undefined

- some expressions denote no normal value (eg *infinity*, $1 / 0$)
- for simplicity (every syntactically well-formed expression denotes a value), introduce special value *undefined* (sometimes written ' \perp ')
 - in evaluating such an expression, evaluator may hang or may give error message
- can apply functions to \perp ; *strict* functions (*square*) give \perp as a result, *nonstrict* functions (*three*) may give some non- \perp value

1.3 Functions

- naturally, FP is a matter of functions
- script defines *functions* (*square*, *smaller*)
- (script actually defines *values*; indeed, in FP functions are values)
- function transforms (one or more) arguments into result
- *deterministic*: same arguments always give same result
- may be *partial*: result may sometimes be \perp
- eg cosine, square root; distance between two cities; compiler; text formatter; process controller

1.3 Function types

- *type declaration* in script specifies type of function
- eg $\text{square} :: \text{Integer} \rightarrow \text{Integer}$
- in general, $f :: A \rightarrow B$ indicates that function f takes arguments of type A and returns results of type B
- *apply* function to argument: $f\ x$
- sometimes parentheses are necessary: $\text{square}\ (3 + 4)$
(function application is an operator, binding more tightly than $+$)
- be careful not to confuse the function f with the value $f\ x$

1.3 Lambda

- notation for anonymous functions
- eg $\lambda x \rightarrow x * x$ as another way of writing *square*
- eg $\lambda a \ b \rightarrow a$ (which we'll call *const* later)
- ASCII ' \backslash ' is nearest equivalent to Greek λ
- from Church's λ -calculus theory of computability (1941)

1.3 Declaration vs expression style

- Haskell is a committee language
- Haskell supports two different programming styles
- *declaration style*: using equations, patterns and expressions

```
quad :: Integer → Integer  
quad x = square x * square x
```

- *expression style*: emphasising the use of expressions

```
quad :: Integer → Integer  
quad = λx → square x * square x
```

- expression style is often more flexible
- experienced programmers use both simultaneously

1.3 Extensionality

- two functions are equal ($f = g$) if they give equal results for all arguments ($f\ x = g\ x$ for every x of the right type)
- this is why the two definitions of *quad* (see previous slide) are equivalent
- the important thing about a function is its mapping from arguments to results
- other properties (eg how a mapping is described) are irrelevant
- eg these two functions are equal, as well:

double, double' :: Integer → Integer

double x = x + x

*double' x = 2 * x*

1.3 Currying

- replace single structured argument by several simpler ones

$add :: (Integer, Integer) \rightarrow Integer$
 $add\ x\ y = x + y$

$add' :: Integer \rightarrow (Integer \rightarrow Integer)$
 $add'\ x\ y = x + y$

- useful for reducing number of parentheses
- add takes a pair of *Integers* and returns an *Integer*
- add' takes an *Integer* and returns a function of type $Integer \rightarrow Integer$
- eg $add'\ 3$ is a function; $(add'\ 3)\ 4$ reduces to 7
- can be written just $add'\ 3\ 4$ (see why shortly)

1.3 Operators

- functions with alphabetic names are *prefix*: $f\ 3\ 4$
- functions with symbolic names are *infix*: $3 + 4$
- make an alphabetic name infix by enclosing in backquotes:
 $17\ 'mod'\ 10$
- make symbolic operator prefix (and curried) by enclosing it in parentheses: $(+)\ 3\ 4$
- thus, $add' = (+)$
- extend notion to include one argument too: *sectioning*
- eg $(1/)$ is the reciprocal function, (>0) is the positivity test

1.3 Associativity

- why operators at all? why not prefix notation?
- there is a problem of ambiguity:

$$x \otimes y \otimes z$$

what does this mean: $(x \otimes y) \otimes z$ or $x \otimes (y \otimes z)$?

- sometimes it doesn't matter, eg addition

$$(x + y) + z = x + (y + z)$$

the operator $+$ is associative

- the operator $+$ has also a neutral element

$$x + 0 = x = 0 + x$$

- 0 and $+$ form a monoid (more later)

1.3 Association

- some operators are not associative ($-$, $/$, \uparrow)
- to disambiguate without parentheses, operators may *associate* to the left or to the right
- eg subtraction associates to the left: $5 - 4 - 2 = -1$
- function application associates to the left: $f\ a\ b$ means $(f\ a)\ b$
- function type operator associates to the right:
 $Integer \rightarrow Integer \rightarrow Integer$ means
 $Integer \rightarrow (Integer \rightarrow Integer)$

1.3 Precedence

- association does not help when operators are mixed

$$x \oplus y \otimes z$$

what does this mean: $(x \oplus y) \otimes z$ or $x \oplus (y \otimes z)$?

- to disambiguate without parentheses, there is a notion of *precedence* (binding power)
- eg $*$ has higher precedence (binds more tightly) than $+$

infixl 7 $*$

infixl 6 $+$

- function application can be seen as an operator, and has the highest precedence, so *square* $3 + 4 = 13$

1.3 Composition

- glue functions together with *function composition*
- defined as follows:

$$\begin{aligned}(\circ) &:: (Integer \rightarrow Integer) \rightarrow (Integer \rightarrow Integer) \\ &\quad \rightarrow (Integer \rightarrow Integer) \\ f \circ g &= \lambda x \rightarrow f(g\ x)\end{aligned}$$

- eg function $square \circ double$ takes 3 to 36
- equivalent definition: $(\circ) \ f\ g\ x = f(g\ x)$
- associative, so parentheses not needed in $f \circ g \circ h$
- (actually has a different type; explained later)

1.4 Definitions

- we've seen some simple definitions of functions so far
- can also define other kinds of values:

```
name :: String  
name = "Jeremy"
```

- all so far have had an identifier (and perhaps formal parameters) on the left, and an expression on the right
- other forms possible: conditional, pattern-matching and local definitions
- also recursive definitions (later sections)

1.4 Conditional definitions

- earlier definition of *smaller* used a *conditional expression*:

smaller :: (*Integer*, *Integer*) → *Integer*
smaller (*x*, *y*) = **if** *x* ≤ *y* **then** *x* **else** *y*

- could also use *guarded equations*:

smaller :: (*Integer*, *Integer*) → *Integer*
smaller (*x*, *y*)
 | *x* ≤ *y* = *x*
 | *x* > *y* = *y*

- each *clause* has a *guard* and an *expression* separated by =
- last guard can be *otherwise* (synonym for *True*)
- especially convenient with three or more clauses
- *declaration style*: guard; *expression style*: **if ... then ... else...**

1.4 Pattern matching

- define function by several equations
- arguments on lhs not just variables, but *patterns*
- patterns may be *variables* or *constants* (or *constructors*, later)
- eg

```
day :: Integer → String  
day 1 = "Saturday"  
day 2 = "Sunday"  
day _ = "Weekday"
```

- also *wildcard pattern* `_`
- evaluate by reducing argument to normal form, then applying first matching equation
- result is \perp if argument has no normal form, or no equation matches

1.4 Local definitions

- repeated subexpression can be captured in a *local definition*

```

roots :: (Float, Float, Float) → (Float, Float)
roots (a, b, c) = ((-b - sd) / (2 * a), (-b + sd) / (2 * a))
  where sd = sqrt (b * b - 4 * a * c)

```

- scope of ‘where’ clause extends over whole right-hand side
- multiple local definitions can be made:

```

demo :: Integer → Integer → Integer
demo x y = (a + 1) * (b + 2)
  where a = x - y
        b = x + y

```

(nested scope, so layout rule applies here too: all definitions must start in same column)

- in conjunction with guarded equations, the scope of a **where** clause covers all guard clauses

1.4 **let**-expressions

- a **where** clause is syntactically attached to an equation
- also: definitions local to an expression

```
demo :: Integer → Integer → Integer  
demo x y = let a = x - y  
           b = x + y  
           in (a + 1) * (b + 2)
```

- *declaration style*: **where**; *expression style*: **let ... in...**
- **let**-expressions are more flexible than **where** clauses

1.5 The art of functional programming

- a problem is given by an expression
- a solution is a value
- a solution is obtained by evaluating an expression to a value
- a program introduces vocabulary to express problems and specifies rules for evaluating expressions
- the art of functional programming: finding rules
- Haskell has a very simple computational model
- as in primary school: replacing equals by equals
- we can calculate not only with numbers, but also with lists, trees, pictures, music ...

Part 2

Types and polymorphism

2.0 Outline

Strong typing

Simple types

Enumerations

Tuples

Polymorphism

Type synonyms

Type classes

Summary

2.1 Strong typing

- Haskell is *strongly typed*: every expression has a unique type
- each type supports certain operations, which are meaningless on other types
- type checking guarantees that type errors cannot occur
- Haskell is *statically typed*: type checking occurs before runtime (after syntax checking)
- experience shows well-typed programs are likely to be correct
- Haskell can *infer types*: determine the most general type of each expression
- wise to specify (some) types anyway, for documentation and redundancy

2.2 Simple types

- booleans
- characters
- strings
- numbers

2.2 Booleans

- type *Bool* (note: type names capitalized)
- two constants, *True* and *False* (note: constructor names capitalized)
- eg definition by pattern-matching

not :: *Bool* → *Bool*

not False = *True*

not True = *False*

- and *&&*, or *||*, both strict in first argument

(&&) :: *Bool* → *Bool* → *Bool*

False && _ = *False*

True && x = *x*

- comparisons *==*, *≠*, orderings *<*, *≤* etc

2.2 Boole pattern

- every type comes with a pattern of definition
- *task*: define a function $f :: Bool \rightarrow S$;
- *step 1*: solve the problem for *False*

$f\ False = \dots$

- *step 2*: solve the problem for *True*

$f\ False = \dots$

$f\ True = \dots$

- (exercise: define your own conditional)

2.2 Characters

- type *Char*
- constants in single quotes: 'a', '?'
- special characters escaped: '\\', '\\n', '\\\\'
- ASCII coding: *ord*:: *Char* → *Int*, *chr*:: *Int* → *Char* (defined in module *Data.Char*)
- comparison and ordering, as before

2.2 Strings

- type *String*
- (actually defined in terms of *Char*; see later)
- constants in double quotes: "Hello"
- comparison and (lexicographic) ordering
- concatenation ++
- display function $\text{show} :: \text{Integer} \rightarrow \text{String}$ (actually more general than this; see later)

2.2 Numbers

- fixed-size (32-bit) integers *Int*
- arbitrary-precision integers *Integer*
- single- and double-precision floats *Float*, *Double*
- others too: rationals, complex numbers, ...
- comparisons and ordering
- $+$, $-$, $*$, \uparrow
- *abs*, *negate*
- $/$, *div*, *mod*, *quot*, *rem*
- etc
- operations are overloaded (more later)

2.3 Enumerations

- mechanism for declaring new types

```
data Day = Mon | Tue | Wed | Thu | Fri | Sat | Sun
```

- eg *Bool* is not built in (although **if ... then ... else** syntax is):

```
data Bool = False | True
```

- types may even be parameterized and recursive! (more later)

2.4 Tuples

- pairing types: eg $(Char, Integer)$
- values in the same syntax: $('a', 440)$
- selectors fst , snd
- definition by pattern matching:

$$fst(x, _) = x$$

- nested tuples: $(Integer, (Char, Bool))$
- triples, etc: $(Integer, Char, Bool)$
- nullary tuple $()$
- comparisons, (lexicographic) ordering

2.5 Polymorphism

- what is the type of *fst*?
- applicable at different types: *fst* (1, 2), *fst* ('a', True), ...
- what about strong typing?
- *fst* is *polymorphic* — it works for *any* type of pairs:

$$fst :: (a, b) \rightarrow a$$

- *a*, *b* here are *type variables* (uncapitalized)
- values can be polymorphic too: $\perp :: a$
- regain principal types for all expressions

2.5 A little game

- here is a little game: I give you a type, you give me a function of that type
 - ▶ $Int \rightarrow Int$
 - ▶ $a \rightarrow a$
 - ▶ $(Int, Int) \rightarrow Int$
 - ▶ $(a, a) \rightarrow a$
 - ▶ $(a, b) \rightarrow a$
 - ▶ $[a] \rightarrow [a]$
- polymorphic functions: flexible to use, hard to define
- polymorphism is a property of an algorithm

2.6 Type synonyms

- alternative names for types
- brevity, clarity, documentation
- eg

type *Card* = (*Rank*, *Suit*)

- cannot be recursive
- just a ‘macro’: no new type

2.7 Type classes

- what about numeric operations?
- $(+) :: Integer \rightarrow Integer \rightarrow Integer$
- $(+) :: Float \rightarrow Float \rightarrow Float$
- cannot have $(+) :: a \rightarrow a \rightarrow a$ (too general)
- the solution is *type classes* (sets of types)
- eg the type class *Num* is a set of numeric types; includes *Integer*, *Float*, etc
- now $(+) :: (Num\ a) \Rightarrow (a \rightarrow a \rightarrow a)$
- *ad hoc polymorphism* (different code for different types), as opposed to *parametric polymorphism* (same code for all types)

2.7 Some standard type classes

- *Eq*: $=$, \neq
- *Ord*: $<$ etc, *min* etc
- *Enum*: *succ*, ..
- *Bounded*: *minBound*, *maxBound*
- *Show*: *show* :: $a \rightarrow \text{String}$
- *Read*: *read* :: $\text{String} \rightarrow a$
- *Num*: $+$, $*$ etc
- *Real* (ordered numeric types)
- *Integral*: *div* etc
- *Fractional*: $/$ etc
- *Floating*: *exp* etc
- more later

2.7 Derived type classes

- new **data**types not automatically instances of useful type classes
- possible to install as instances:

instance *Eq Day* **where**

Mon == *Mon* = *True*

Tue == *Tue* = *True*

Wed == *Wed* = *True*

Thu == *Thu* = *True*

Fri == *Fri* = *True*

Sat == *Sat* = *True*

Sun == *Sun* = *True*

_ == *_* = *False*

- (default definition of \neq obtained for free from `==`, more later)
- tedious for simple cases, which can be derived automatically:

data *Day* = *Mon* | *Tue* | *Wed* | *Thu* | *Fri* | *Sat* | *Sun*

deriving (*Eq, Ord, Enum, Bounded, Show, Read*)

2.8 Type-driven program development

- types are a vital part of any program
- types are not an afterthought
- first specify the type of a function
- its definition is then driven by the type

$$f :: T \rightarrow U$$

- f consumes a T value: suggests case analysis
- f produces a U value: suggests use of constructors
- type safety and flexibility are in tension
- polymorphism partially releases the tension

Part 3

Lists

3.0 Outline

List notation

Compositional programming

List constructors

List definition pattern

Some list operations

List comprehensions

Summary

3.1 List notation

- lists are central to functional programming (cf LISP!)
- sequences of elements of the same type
- enclosed in square brackets, comma-separated: `[1, 2, 3], []`
- the type of lists with elements of type *a* is `[a]`
- strings are just lists of characters: `['H', 'e', 'l', 'l', 'o']`

`type String = [Char]`

but with special syntax `"Hello"`

- list elements can be any type:

```
[1, 2, 3]           :: [Integer]
[[1, 2], [], [3]] :: [[Integer]]
[(+), (*)]         :: [Integer → Integer → Integer]
```

3.2 Some library functions

- exploring the library *Data.List*

import *Data.List*

- concat* :: $[[a]] \rightarrow [a]$ **eg** *concat* $[[1,2], [], [3]] = [1,2,3]$
- length* :: $[a] \rightarrow \text{Int}$ **eg** *length* $[1,2,3] = 3$
- reverse* :: $[a] \rightarrow [a]$ **eg** *reverse* "jeremy" = "ymerej"
- map* :: $(a \rightarrow b) \rightarrow ([a] \rightarrow [b])$ **eg** *map* $(+1) [1,2,3] = [2,3,4]$
- lines* :: *String* $\rightarrow [\text{String}]$ **eg**
lines "a\nbc\nd" = $["a", "bc", "d"]$
- unlines* :: $[\text{String}] \rightarrow \text{String}$ **eg**
unlines $["a", "bc", "d"] = "a\nbc\nd\n"$
- tails* :: $[a] \rightarrow [[a]]$ **eg** *tails* "jeremy" =
 $["jeremy", "eremy", "remy", "emy", "my", "y", ""]$

3.2 How to solve it?

- write down the type (what's the input?, what's the output?)
- can you solve it using existing vocabulary?
- use function application and function composition
- some exercises: given a string (a list of characters)
 - ▶ remove newlines
 - ▶ count the number of lines
 - ▶ flip text upside down
 - ▶ flip text from left to right
 - ▶ determine the list of all substrings

3.2 Solutions

- remove newlines

unwrap :: String → String
unwrap = concat ∘ lines

- count the number of lines

countLines :: String → Int
countLines = length ∘ lines

- flip text upside down

upsideDown :: String → String
upsideDown = unlines ∘ reverse ∘ lines

- flip text from left to right

leftRight :: String → String
leftRight = unlines ∘ map reverse ∘ lines

3.2 Solutions continued

- determine the list of all prefixes (actually, also defined in the library: *inits*)

```
suffixes, prefixes :: String → [String]  
suffixes = tails  
prefixes = map reverse ∘ tails ∘ reverse
```

- determine the list of all substrings

```
substrings :: String → [String]  
substrings = concat ∘ map prefixes ∘ suffixes
```

3.3 List constructors

- a list is either
 - empty, written `[]`
 - or consists of an element `x` followed by a list `xs`, written `x:xs`
- every finite list can be built up from `[]` using `:`
- eg `[1,2,3] = 1:(2:(3:[])) = 1:2:3:[]`
- `[]` and `:` are called *constructors*

3.3 Type of list constructors

- *nil*: the empty list

$$[] :: [a]$$

- *cons*: function for prefixing an element onto a list

$$(:) :: a \rightarrow [a] \rightarrow [a]$$

- `[]` and `:` are polymorphic!
- puzzle: is `[]:[]` well-typed? what about `[]:([]:[])` and `([]:[]):[]`?

3.4 Pattern matching

- constructors are exhaustive
- to define function over lists, it suffices to consider the two cases `[]` and `:`
- eg to test if list is empty

$$\begin{aligned} \text{null} &:: [a] \rightarrow \text{Bool} \\ \text{null } [] &= \text{True} \\ \text{null } (x:xs) &= \text{False} \end{aligned}$$

(why is this different from `(== [])`?)

- eg to return first element of non-empty list

$$\begin{aligned} \text{head} &:: [a] \rightarrow a \\ \text{head } (x:xs) &= x \end{aligned}$$

3.4 Case analysis

- cases can also be analysed using a **case**-expression

```
null :: [ a ] → Bool  
null xs = case xs of  
    [ ]      → True  
    ( _ : _ ) → False
```

- declaration style*: equation using patterns; *expression style*: **case**-expression using patterns

3.4 Recursive definitions

- definitions by pattern-matching can be recursive too
- natural as the type is also recursively defined
- eg sum of a list of integers

```
sum :: [ Integer ] → Integer  
sum [ ]      = 0  
sum (x : xs) = x + sum xs
```

- eg length of a list of elements

```
length :: [ a ] → Int  
length [ ]      = 0  
length (x : xs) = 1 + length xs
```

3.4 List definition pattern

- remember: every type comes with a definition pattern
- *task*: define a function $f :: [P] \rightarrow S$
- *step 1*: solve the problem for the empty list

$$f [] = \dots$$

- *step 2*: solve the problem for the non-empty list;
assume that you already have the solution for xs at hand;
extend the intermediate solution to a solution for $x : xs$

$$\begin{aligned} f [] &= \dots \\ f (x : xs) &= \dots x \dots xs \dots f xs \dots \end{aligned}$$

you have to program only a *step*

- put on your problem-solving glasses

3.5 Some list operations

- **append:** $[1, 2, 3] \mathbin{++} [4, 5] = [1, 2, 3, 4, 5]$

$$(\mathbin{++}) :: [a] \rightarrow [a] \rightarrow [a]$$

$$[] \mathbin{++} ys = ys$$

$$(x:xs) \mathbin{++} ys = x: (xs \mathbin{++} ys)$$

- **concatenation:** $concat\ [[1, 2], [], [3]] = [1, 2, 3]$

$$concat :: [[a]] \rightarrow [a]$$

$$concat\ [] = []$$

$$concat\ (x:xs) = x \mathbin{++} concat\ xs$$

- **reverse:** $reverse\ [1, 2, 3] = [3, 2, 1]$

$$reverse :: [a] \rightarrow [a]$$

$$reverse\ [] = []$$

$$reverse\ (x:xs) = reverse\ xs \mathbin{++} [x]$$

(exercise: complexity? improve!)

- is a list ordered?

ordered :: (Ord a) ⇒ [a] → Bool

ordered [] = True

ordered [x] = True

ordered (x:y:xs) = x ≤ y && *ordered* (y:xs)

we distinguish three cases

- zip: eg *zip* [1,2,3] "ab" = [(1, 'a'), (2, 'b')]

zip :: [a] → [b] → [(a,b)]

zip [] [] = []

zip [] (_:_) = []

zip (_:_) [] = []

zip (x:xs) (y:ys) = (x,y) : *zip* xs ys

we pattern match on both arguments

3.6 List comprehensions

- two useful operators on lists: *map* and *filter*
- list comprehensions provide a convenient syntax for expressions involving *map*, *filter*, *concat*
- analogous to a database query language
- useful for constructing new lists from old lists

3.6 Map

- applies given function to every element of given list
- eg *map square* [1, 2, 3] = [1, 4, 9]
- eg *map succ* "HAL" = "IBM"
- definition

$$\text{map} :: (a \rightarrow b) \rightarrow ([a] \rightarrow [b])$$
$$\text{map } f [] = []$$
$$\text{map } f (x:xs) = f x : \text{map } f xs$$

- another eg: *sum (map square [1..10])*
- (special syntax [*m..n*] for enumerations)

3.6 Filter

- returns sublist of the argument whose elements satisfy given predicate
- eg *filter isDigit "more4u2say" = "42"*
- eg *(sum ◦ map square ◦ filter odd) [1..5] = 35*
- definition

filter :: (*a* → *Bool*) → ([*a*] → [*a*])

filter p [] = []

filter p (x:xs)

| *p x* = *x: filter p xs*

| *otherwise* = *filter p xs*

3.6 Comprehensions

- special convenient syntax for list-generating expressions
- eg *sum* [*square* *x* | *x* ← [1..5], *odd* *x*]
- formally, a comprehension [*e* | *Qs*] for expression *e* and non-empty comma-separated sequence of qualifiers *Qs*
- qualifier may be *generator* (of the form *x* ← *xs*) or *guard* (a boolean expression)

3.6 Examples of comprehensions

- eg primes up to a given bound

```
primes, divisors :: Integer → [ Integer ]  
primes m = [ n | n ← [ 1 .. m ], divisors n == [ 1, n ] ]  
divisors n = [ d | d ← [ 1 .. n ], n 'mod' d == 0 ]
```

- eg database query

```
overdue =  
  [ ( name, addr ) | ( key, name, addr ) ← names,  
                    ( key', date, amount ) ← invoices, key == key',  
                    date < today ]
```

- eg Quicksort

```
quicksort :: (Ord a) ⇒ [ a ] → [ a ]  
quicksort [ ]      = [ ]  
quicksort (x : xs) = quicksort [ y | y ← xs, y < x ]  
                  ++ [ x ] ++  
                  quicksort [ y | y ← xs, y ≥ x ]
```

3.6 Another point of view

- list comprehension is ‘really’ a form of nested loop
- eg $[f\ b \mid a \leftarrow x, b \leftarrow g\ a, p\ b]$ is related to

```
foreach (int a in x)
  foreach (int b in g a)
    if (p b)
      yield (f b)
```

3.6 Advanced: semantics by translation ★

- generator iterates over list, binding new variable

$$[e \mid x \leftarrow xs, Qs] = \text{concat} (\text{map} (\lambda x \rightarrow [e \mid Qs]) xs)$$

- guard prunes collection

$$[e \mid p, Qs] = \text{if } p \text{ then } [e \mid Qs] \text{ else } []$$

- empty qualifier list generates a singleton

$$[e \mid] = [e]$$

- eg

$$\begin{aligned} & [x * x \mid x \leftarrow [1..5], \text{odd } x] \\ = & \text{concat} (\text{map} (\lambda x \rightarrow [x * x \mid \text{odd } x]) [1..5]) \\ = & \text{concat} (\text{map} (\lambda x \rightarrow \text{if } \text{odd } x \text{ then } [x * x] \text{ else } []) [1..5]) \\ = & \text{concat} [[1 * 1], [], [3 * 3], [], [5 * 5]] \\ = & [1, 9, 25] \end{aligned}$$

3.7 Summary: How to solve it?

- write down the type (what's the input?, what's the output?)
- can you solve the problem using existing vocabulary?
- if not, define new vocabulary
- use the list definition pattern
- remember: you only have to solve a step
- can you solve the step using existing vocabulary?
- if not, define new vocabulary (identify a subproblem)
- solve the subproblem in the same manner

Part 4

Algebraic datatypes

4.0 Outline

New datatypes

Product and sum datatypes

Parametric datatypes

Recursive datatypes

Case study: compiler construction

Summary

4.1 New datatypes

- we've seen **type** synonyms for existing types
- we've also seen enumerations as new **data** types
- **data** is *much* more general than this
- product and sum datatypes
- parametric datatypes
- recursive datatypes

4.2 Product and sum datatypes

- constructors of enumerated types are constants (*Mon*); constructors may be functions too
- eg people with names and ages

```
type Name = String
type Age = Int
data Person = P Name Age
```

- then $P :: \text{Name} \rightarrow \text{Age} \rightarrow \text{Person}$
- such *constructor functions* do not simplify, they are in normal form; moreover, they can be used in pattern-matching

```
showPerson :: Person → String
showPerson (P n a) = "Name: " ++ n ++ ", Age: " ++ show a
```

- safer than type synonyms, and can have their own type classes (eg specialized equality)

```
type Person = (Name, Age)
```

4.2 Sum types

- datatypes can have multiple *variants*

data *Suit* = *Spades* | *Hearts* | *Diamonds* | *Clubs*

data *Rank* = *Faceless Integer* | *Jack* | *Queen* | *King*

data *Card* = *Card Rank Suit* | *Joker*

- so a *Rank* is *either* of the form *Faceless n* for some *n*, *or* a constant *Jack*, *Queen* or *King*

4.2 Temperatures

- another example

data *Temp* = *Cels Float* | *Fahr Float*
deriving (*Show*)

- define our own equality function

instance *Eq Temp* **where**
 Cels *x* == *Cels* *y* = *x* == *y*
 Fahr *x* == *Fahr* *y* = *x* == *y*
 Cels *x* == *Fahr* *y* = *x* * 1.8 == *y* - 32.0
 Fahr *x* == *Cels* *y* = *Cels* *y* == *Fahr* *x*

4.3 Parametric datatypes

- constructors may be polymorphic functions
- then datatype is parametric

data *Maybe a = Just a | Nothing*

- eg *Just 13 :: Maybe Int*
- so *Just :: a → Maybe a*, *Nothing :: Maybe a*
- useful for modelling exceptions

head' :: [a] → Maybe a
head' [] = Nothing
head' (x:_) = Just x

- similarly, sum datatype

data *Either a b = Left a | Right b*

4.4 Recursive datatypes

- datatypes may be recursive too
- arithmetic expressions
- natural numbers
- lists
- binary trees
- general trees

4.4 Arithmetic expressions

- datatype of arithmetic expressions

data *Expr* = *Lit Integer* | *Add Expr Expr* | *Mul Expr Expr*

- an arithmetic expressions is either a literal, or two expressions added together, or two multiplied
- constructor names may be operators (starting with ':')

infixl 7 ::

infixl 6 ::

data *Expr*

 = *Lit Integer* -- a literal

 | *Expr* :: *Expr* -- addition

 | *Expr* :: *Expr* -- multiplication

deriving (*Show*)

4.4 Constructing expressions

- constructing expressions

expr1, expr2 :: Expr

*expr1 = (Lit 4 *: Lit 7) :+: (Lit 11)*

*expr2 = (Lit 4 :+: Lit 7) *: (Lit 11)*

- note the difference between *syntax*

*? Lit 4 :+: Lit 7 *: Lit 11*

*Lit 4 :+: Lit 7 *: Lit 11*

- and *semantics*

*? 4 + 7 * 11*

81

4.4 *Expr* definition pattern

- recursive definitions by pattern-matching

evaluate :: *Expr* → *Integer*

evaluate (*Lit* *i*) = *i*

evaluate (*e1* :+ : *e2*) = *evaluate e1* + *evaluate e2*

evaluate (*e1* :* : *e2*) = *evaluate e1* * *evaluate e2*

- the evaluator essentially replaces syntax (:+ : and :* :) by semantics (+ and *)

4.4 *Expr* definition pattern

- remember: every datatype comes with a definition pattern
- task*: define a function $f :: Expr \rightarrow S$
- step 1*: solve the problem for literals

$$f(Lit\ n) = \dots n \dots$$

- step 2*: solve the problem for addition;
assume that you already have the solution for x and y at hand;
extend the intermediate solution to a solution for $x :+: y$

$$f(Lit\ n) = \dots$$

$$f(x :+: y) = \dots x \dots y \dots f\ x \dots f\ y \dots$$

you have to program only a *step*

- step 3*: do the same for $x **: y$

$$f(Lit\ n) = \dots$$

$$f(x :+: y) = \dots x \dots y \dots f\ x \dots f\ y \dots$$

$$f(x **: y) = \dots x \dots y \dots f\ x \dots f\ y \dots$$

4.4 Naturals

- *Peano* definition of natural numbers (non-negative integers)

data *Nat* = *Zero* | *Succ Nat*

- every natural is either *Zero* or the *Successor* of a natural
- eg *Succ (Succ (Succ Zero))* corresponds to 3
- extraction

nat2int :: *Nat* → *Integer*

nat2int Zero = 0

nat2int (Succ n) = 1 + *nat2int n*

- addition

plus :: *Nat* → *Nat* → *Nat*

plus Zero n = *n*

plus (Succ m) n = *Succ (plus m n)*

(does this look familiar?)

4.4 Peano definition pattern

- remember: every datatype comes with a definition pattern
- task*: define a function $f :: Nat \rightarrow S$
- step 1*: solve the problem for *Zero*

$$f \textit{Zero} = \dots$$

- step 2*: solve the problem for *Succ n*;
assume that you already have the solution for *n* at hand;
extend the intermediate solution to a solution for *Succ n*

$$\begin{aligned} f \textit{Zero} &= \dots \\ f (\textit{Succ } n) &= \dots n \dots f n \dots \end{aligned}$$

you have to program only a *step*

- put on your problem-solving glasses
- (exercise: *n*th power)

4.4 Lists

- built-in type of lists is not special (has only special syntax)

data *List a = Nil | Cons a (List a)*

- eg `[1,2,3]` or `1:2:3:[]` corresponds to
Cons 1 (Cons 2 (Cons 3 Nil))
- recursive definitions by pattern-matching

mapList :: (a → b) → (List a → List b)

mapList f Nil = Nil

mapList f (Cons x xs) = Cons (f x) (mapList f xs)

4.4 List definition pattern

- remember: every datatype comes with a definition pattern
- *task*: define a function $f :: List\ P \rightarrow S$
- *step 1*: solve the problem for the empty list

$$f\ Nil = \dots$$

- *step 2*: solve the problem for the non-empty list;
assume that you already have the solution for xs at hand;
extend the intermediate solution to a solution for $Cons\ x\ xs$

$$\begin{aligned} f\ Nil &= \dots \\ f\ (Cons\ x\ xs) &= \dots\ x\ \dots\ xs\ \dots\ f\ xs\ \dots \end{aligned}$$

you have to program only a *step*

- put on your problem-solving glasses

4.4 Binary trees

- externally-labelled binary trees

data *Btree* *a* = *Tip* *a* | *Bin* (*Btree* *a*) (*Btree* *a*)

- eg *Bin* (*Tip* 1) (*Bin* (*Tip* 2) (*Tip* 3))
- eg *size* (number of elements)

size :: *Btree* *a* → *Int*

size (*Tip* *x*) = 1

size (*Bin* *t* *u*) = *size* *t* + *size* *u*

4.4 General trees

- internally-labelled trees with arbitrary branching (*rose trees*)

data *Gtree* *a* = *Branch* *a* [*Gtree* *a*]

- eg
Branch 1 [*Branch* 2 [], *Branch* 3 [*Branch* 4 []], *Branch* 5 []]
- eg given available moves $m :: Pos \rightarrow [Pos]$, generate game tree

gametree :: $(Pos \rightarrow [Pos]) \rightarrow (Pos \rightarrow Gtree\ Pos)$
gametree *m* *p* = *Branch* *p* (map (*gametree* *m*) (*m* *p*))

4.5 Case study: compiler construction

- let's implement a compiler that translates arithmetic expressions into stack machine code and
- a virtual machine that executes stack machine code

compile (Lit 4 :: (Lit 7 :+: Lit 11))
= Push 4 :^: Push 7 :^: Push 11 :^: Add :^: Mul*

- when executed, the stack grows and shrinks

*[]
4 : []
7 : 4 : []
11 : 7 : 4 : []
18 : 4 : []
72 : []*

- we also show the correctness of compiler and VM

4.5 Warm-up: showing expressions

- *showExpr* maps an expression to its string representation

showExpr :: *Expr* → *String*

showExpr (*Lit i*)

 = *show i*

showExpr (*e1* :+: *e2*)

 = "(" ++ *showExpr e1* ++ " + " ++ *showExpr e2* ++ ")"

showExpr (*e1* **: *e2*)

 = "(" ++ *showExpr e1* ++ " * " ++ *showExpr e2* ++ ")"

- parentheses is necessary for products of sums eg

showExpr expr2 = "((4 + 7) * 11)"

- some parentheses is redundant, however, eg

showExpr expr1 = "(4 * 7) + 11)"

4.5 Respecting precedence

- string representation should respect precedence

```
infixl 7 :*:  
infixl 6 :+:
```

- *idea*: pass in the precedence level of the enclosing operator

```
showPrec :: Int → Expr → String
```

```
showPrec _ (Lit i)  
  = show i
```

```
showPrec p (e1 :+: e2)
```

```
  = parenthesis (p > 6) (showPrec 6 e1 ++ " + " ++ showPrec 6 e2)
```

```
showPrec p (e1 :*: e2)
```

```
  = parenthesis (p > 7) (showPrec 7 e1 ++ " * " ++ showPrec 7 e2)
```

4.5 Respecting precedence—continued

- optional parenthesis

```
parenthesis :: Bool → String → String  
parenthesis True s = "(" ++ s ++ ")"  
parenthesis False s = s
```

- we start off with the lowest precedence

```
showExpr :: Expr → String  
showExpr = showPrec 0
```

- eg *showExpr* *expr1* = "4 * 7 + 11" and
showExpr *expr2* = "(4 + 7) * 11"

4.5 Instructions of a stack machine

- the operations of the VM operate on a stack

infixr 2 :[^]:

data *Code*

 = *Push Integer* -- push integer onto stack
 | *Add* -- add topmost two elements and push result
 | *Mul* -- multiply
 | *Code* :[^] : *Code* -- sequencing

deriving (*Show*)

- eg

code1 :: *Code*

code1 = *Push* 47 :[^] : *Push* 11 :[^] : *Add*

4.5 Warm-up: showing code

- *showCode* maps a piece of code to its string representation

showCode :: *Code* → *String*

showCode (*Push* *i*) = "push " ++ *show* *i*

showCode (*Add*) = "add"

showCode (*Mul*) = "mul"

showCode (*c1* : ^ : *c2*) = *showCode* *c1* ++ " ; " ++ *showCode* *c2*

- eg *showCode* *code1* = "push 47 ; push 11 ; add"

4.5 Compilation

- the compiler follows the *Expr* definition pattern

compile :: *Expr* → *Code*

compile (*Lit* *i*) = *Push* *i*

compile (*e1* :+ *e2*) = *compile* *e1* :^: *compile* *e2* :^: *Add*

compile (*e1* :* *e2*) = *compile* *e1* :^: *compile* *e2* :^: *Mul*

- for addition we first generate code for the two subexpressions and then emit an *Add* instruction
- eg *compile* *expr1* = *Push* 4 :^: *Push* 7 :^: *Mul* :^: *Push* 11 :^: *Add*

4.5 Execution

- we implement a stack using a list of integers

type *Stack* = [*Integer*]

- the VM follows the *Code* definition pattern

execute :: *Code* → (*Stack* → *Stack*)
execute (*Push* *i*) = *push* *i*
execute (*Add*) = *add*
execute (*Mul*) = *mul*
execute (*c1* :[^]: *c2*) = *execute* *c2* ∘ *execute* *c1*

- syntax (*Push*) is replaced by semantics (*push*)

4.5 Helper functions

- push* etc are *stack transformers*

push :: *Integer* → (*Stack* → *Stack*)

push *i* *xs* = *i* : *xs*

add :: *Stack* → *Stack*

add [] = *error msg*

add [_] = *error msg*

add (*x1* : *x2* : *xs*) = *x2* + *x1* : *xs*

mul :: *Stack* → *Stack*

mul [] = *error msg*

mul [_] = *error msg*

mul (*x1* : *x2* : *xs*) = *x2* * *x1* : *xs*

msg :: *String*

msg = "VM: empty stack"

4.5 Advanced: proof of correctness ★

Executing a compiled expression has the same effect as evaluating the expression and then pushing the result:

$$\textit{push}(\textit{evaluate } e) = \textit{execute}(\textit{compile } e)$$

The proof proceeds by induction over the structure of the expression e .

4.5 Proof of correctness: base case ★

Case $e = \text{Lit } i$:

$$\begin{aligned} & \text{execute (compile (Lit } i)) \\ = & \quad \{ \text{definition of compile} \} \\ & \text{execute (Push } i) \\ = & \quad \{ \text{definition of execute} \} \\ & \text{push } i \\ = & \quad \{ \text{definition of evaluate} \} \\ & \text{push (evaluate (Lit } i)) \end{aligned}$$

4.5 Proof of correctness: inductive step ★

Case $e = e1 :+: e2$:

$$\begin{aligned} & \text{execute (compile (e1 :+: e2))} \\ = & \quad \{ \text{definition of compile} \} \\ & \text{execute (compile e1}^{\wedge} \text{: compile e2}^{\wedge} \text{: Add)} \\ = & \quad \{ \text{definition of execute} \} \\ & \text{add} \circ \text{execute (compile e2)} \circ \text{execute (compile e1)} \\ = & \quad \{ \text{induction hypothesis} \} \\ & \text{add} \circ \text{push (evaluate e2)} \circ \text{push (evaluate e1)} \\ = & \quad \{ \text{property of add: } \text{add} \circ \text{push } n \circ \text{push } m = \text{push } (m + n) \} \\ & \text{push (evaluate e1 + evaluate e2)} \\ = & \quad \{ \text{definition of evaluate} \} \\ & \text{push (evaluate (e1 :+: e2))} \end{aligned}$$

Likewise for $e1 :* e2$.

4.6 The art of functional programming

- model static aspects of the real world using datatypes
- model dynamic aspects using functions
- don't shy away from introducing new types

Part 5

Higher-order programming

5.0 Outline

Functions as first-class citizens

Functions as arguments

Functions as results

Functions as datastructures

Fold and unfold

Component-oriented and combinator-style programming

Summary

5.1 Functions as first-class citizens

- *functional programming* concerns functions (of course!)
- functions are first-class citizens of the language
- functions have all the rights of other types:
 - ▶ may be passed as arguments
 - ▶ may be returned as results
 - ▶ may be stored in data structures
 - ▶ etc
- functions that manipulate functions are *higher order*

Slogan: higher-order functions allow new and better means of modularizing programs

5.2 Functions as arguments

- we have already seen many examples of higher-order operators encapsulating patterns of computation:
map, filter, reduce
- each is a parameterizable program scheme
- parameterization improves modularity, and hence understanding, modification and reuse

5.3 Functions as results

- functions may also be returned as results

addOrMul :: *Bool* → (*Integer* → *Integer* → *Integer*)
addOrMul *b* = **if** *b* **then** (+) **else** (*)

- partial application
- currying
- function composition (again)

5.3 Partial application

- consider $\text{add}' x y = x + y$
- type $\text{Integer} \rightarrow \text{Integer} \rightarrow \text{Integer}$; takes two *Integers* and returns an *Integer* (eg $\text{add}' 3 4 = 7$)
- another view: type $\text{Integer} \rightarrow (\text{Integer} \rightarrow \text{Integer})$ (remember, \rightarrow associates to the right); takes a single *Integer* and returns an $\text{Integer} \rightarrow \text{Integer}$ function (eg $\text{add}' 3$ is the *Integer*-transformer that adds three)
- need not apply function to all its arguments at once: *partial application*; result will then be a function, awaiting remaining arguments
- in fact, partial application is the norm; every function takes exactly one argument
- sectioning $((3+), (+))$ is partial application of binary ops

5.3 Currying

- a function taking pair of arguments can be transformed into a function taking two successive arguments, and vice versa

$add :: (Integer, Integer) \rightarrow Integer$

$add\ x\ y = x + y$

$add' :: Integer \rightarrow Integer \rightarrow Integer$

$add'\ x\ y = x + y$

- add' is called the *curried* version of add
- named after logician Haskell B. Curry (like the language), though actually due to Schönfinkel
- thus, pair-consuming functions are unnecessary

- transformations are implementable as higher-order operations

$$\begin{aligned} \text{curry} &:: ((a, b) \rightarrow c) \rightarrow (a \rightarrow b \rightarrow c) \\ \text{curry } f \ a \ b &= f \ (a, b) \end{aligned}$$
$$\begin{aligned} \text{uncurry} &:: (a \rightarrow b \rightarrow c) \rightarrow ((a, b) \rightarrow c) \\ \text{uncurry } f \ (a, b) &= f \ a \ b \end{aligned}$$

- eg $\text{add}' = \text{curry } \text{add}$
- a related higher-order operation: flip arguments of binary function (later: $\text{reverse} = \text{foldl } (\text{flip } (:)) \ []$)

$$\begin{aligned} \text{flip} &:: (a \rightarrow b \rightarrow c) \rightarrow (b \rightarrow a \rightarrow c) \\ \text{flip } f \ b \ a &= f \ a \ b \end{aligned}$$

5.3 Function composition

- recall function composition (now with polymorphic type)

$$\begin{aligned}(\circ) &:: (b \rightarrow c) \rightarrow (a \rightarrow b) \rightarrow a \rightarrow c \\ (f \circ g) \ x &= f (g \ x)\end{aligned}$$

- takes two functions that ‘meet in the middle’ and an argument to one; returns the result from the other
- equivalently, type $(b \rightarrow c) \rightarrow (a \rightarrow b) \rightarrow (a \rightarrow c)$
- takes two functions, glues them together to form a third
- exercise:* show that \circ is associative

5.3 Repeated composition

- double application: eg *twice square* 3 = 81

twice :: $(a \rightarrow a) \rightarrow (a \rightarrow a)$

twice *f* = *f* ∘ *f*

- generalize: eg *iter* 4 (2*) 1 = 2 * 2 * 2 * 2 * 1

iter :: *Integer* → $(a \rightarrow a) \rightarrow (a \rightarrow a)$

iter 0 *f* = *id*

iter *n* *f* = *f* ∘ *iter* (*n* − 1) *f*

- more on this in a minute ...

5.4 Functions as datastructures

consider a dictionary (associative array)

type *Dict k v*

empty :: *Dict k v*

insert :: $(Eq\ k) \Rightarrow (k, v) \rightarrow Dict\ k\ v \rightarrow Dict\ k\ v$

lookup :: $(Eq\ k) \Rightarrow Dict\ k\ v \rightarrow k \rightarrow v$

5.4 Implementation as list

```
type Dict k v = [ (k, v) ]
```

```
empty :: Dict k v
```

```
empty = [ ]
```

```
insert :: (Eq k) => (k, v) -> Dict k v -> Dict k v
```

```
insert kv kvs = kv : kvs
```

```
lookup :: (Eq k) => Dict k v -> k -> v
```

```
lookup [ ] _ = error "item not present"
```

```
lookup ((k, v) : kvs) k'
```

```
  | k == k'      = v
```

```
  | otherwise = lookup kvs k'
```

5.4 Implementation as function

type *Dict* *k v* = *k* → *v*

empty :: *Dict k v*

empty _ = error "item not present"

insert :: (*Eq k*) ⇒ (*k, v*) → *Dict k v* → *Dict k v*

insert (*k, v*) *f k'*

| *k* == *k'* = *v*

| otherwise = *f k'*

lookup :: (*Eq k*) ⇒ *Dict k v* → *k* → *v*

lookup f = *f*

The dictionary is the look-up function.

5.4 Natural numbers as functions

Functions can be used to represent other data structures.
In fact, we've already seen how to represent the natural numbers as functions, via repeated composition.

type *Natural* = $\forall a. (a \rightarrow a) \rightarrow (a \rightarrow a)$

zero :: *Natural*

zero *f* = *id*

succ :: *Natural* \rightarrow *Natural*

succ *n* *f* = *f* \circ *n* *f*

The \forall makes explicit that these functions are polymorphic.
These are called *Church numerals*. We could define:

one, *two* :: *Natural*

one = *succ zero*

two = *succ one*

Conversion from *Integer* using *iter*; how about back again?

5.5 Fold and unfold

- many recursive definitions on lists share a *pattern* of computation
- capture that pattern as a function (abstraction, conciseness, general properties, familiarity, ...)
- *map* and *filter* are two common patterns
- folds and unfolds capture many more

5.5 Fold right

- consider following pattern of definition

$$\begin{aligned} h [] &= e \\ h (x:xs) &= x \text{ 'op' } h \text{ xs} \end{aligned}$$

(simple variant of list definition pattern: xs is only used in the recursive call)

- then

$$h (x: (y: (z: []))) = x \text{ 'op' } (y \text{ 'op' } (z \text{ 'op' } e))$$

- h replaces constructors by functions
- capture pattern as *foldr*

$$\begin{aligned} foldr &:: (a \rightarrow b \rightarrow b) \rightarrow b \rightarrow [a] \rightarrow b \\ foldr \text{ op } e [] &= e \\ foldr \text{ op } e (x:xs) &= x \text{ 'op' } foldr \text{ op } e \text{ xs} \end{aligned}$$

- difference to *reduce*?

5.5 Examples of fold right

- many examples:

```
sum      = foldr (+) 0
copy     = foldr (:) []
length  = foldr (\x n → 1 + n) 0
map f    = foldr ((:) ∘ f) []
concat  = foldr (++) []
reverse = foldr snoc [] where snoc x xs = xs ++ [x]
xs ++ ys = foldr (:) ys xs
```

- right-to-left computation
- operator may (+, ++) but need not (:, snoc) be associative

5.5 Sorting

- given

$insertList :: (Ord\ a) \Rightarrow a \rightarrow [a] \rightarrow [a]$

$insertList\ x\ [] = [x]$

$insertList\ x\ (y:ys)$

$|\ x \leq y \quad =\ x:y:ys$

$| \textit{otherwise} =\ y:insertList\ x\ ys$

- we have

$insertSort :: (Ord\ a) \Rightarrow [a] \rightarrow [a]$

$insertSort = foldr\ insertList\ []$

5.5 Fold left

- not every list function is a *foldr* (eg *drop*)
- even those that are may have better definitions
- eg *decimal* $[1, 2, 3] = 123$
- efficient algorithm using *Horner's rule*:

$$\text{decimal } [x, y, z] = 10 * (10 * (10 * 0 + x) + y) + z$$

- left-to-right computation — hence *foldl*

$$\text{foldl } op \ e \ [x, y, z] = ((e \ 'op' \ x) \ 'op' \ y) \ 'op' \ z$$

- definition

$$\text{foldl} :: (b \rightarrow a \rightarrow b) \rightarrow b \rightarrow [a] \rightarrow b$$

$$\text{foldl } _ \ e \ [] = e$$

$$\text{foldl } op \ e \ (x : xs) = \text{foldl } op \ (e \ 'op' \ x) \ xs$$

5.5 Accumulating parameter

- recall *reverse* program

$$\begin{aligned} \text{reverse} &:: [a] \rightarrow [a] \\ \text{reverse} &= \text{foldr} (\lambda x \text{ xs} \rightarrow \text{xs} \mathbin{++} [x]) [] \end{aligned}$$

- another definition

$$\begin{aligned} \text{reverse}' &:: [a] \rightarrow [a] \\ \text{reverse}' &= \text{foldl} (\text{flip } (:)) [] \end{aligned}$$

- (now what is complexity?)
- second argument of *foldl* is an *accumulating parameter*

5.5 Fold over non-empty lists

- consider computing maximum element in a list of numbers
- $\text{maximum } (x:xs) = x \text{ 'max' } \text{maximum } xs$ suggests *foldr*
- but what is the starting value e ?
- e should be $\text{maximum } []$; also require $x \text{ 'max' } e = x$
- want to choose e to be smallest possible value
- could restrict to context *Bounded*, with $e = \text{minBound}$
- alternatively, don't use on empty list!

- capture pattern via two folds on non-empty lists

$$\begin{aligned} \text{foldr1} &:: (a \rightarrow a \rightarrow a) \rightarrow [a] \rightarrow a \\ \text{foldr1 } op & (x:xs) \\ &| \text{ null } xs \quad = x \\ &| \text{ otherwise} = x 'op' \text{ foldr1 } op xs \end{aligned}$$
$$\begin{aligned} \text{foldl1} &:: (a \rightarrow a \rightarrow a) \rightarrow [a] \rightarrow a \\ \text{foldl1 } op & (x:xs) = \text{foldl } op x xs \end{aligned}$$

- eg

$$\begin{aligned} \text{foldr1 } op [x, y, z] &= x 'op' (y 'op' z) \\ \text{foldl1 } op [x, y, z] &= (x 'op' y) 'op' z \end{aligned}$$

- now $\text{maximum} = \text{foldr1 } \text{max} = \text{foldl1 } \text{max}$

5.5 Scan

- sometimes convenient to apply *foldl* to every initial segment of a list

$$\text{scanl } op \ e \ [x, y, z] = [e, e \ 'op' \ x, (e \ 'op' \ x) \ 'op' \ y, ((e \ 'op' \ x) \ 'op' \ y) \ 'op' \ z]$$

- eg *scanl* (+) 0 computes *prefix sums*
- eg *scanl* (*) 1 [1..n] computes first $n + 1$ factorials
- start with specification

$$\begin{aligned} \text{scanl} &:: (b \rightarrow a \rightarrow b) \rightarrow b \rightarrow [a] \rightarrow [b] \\ \text{scanl } op \ e &= \text{map } (\text{foldl } op \ e) \circ \text{inits} \end{aligned}$$

$$\begin{aligned} \text{inits} &:: [a] \rightarrow [[a]] \\ \text{inits } [] &= [[]] \\ \text{inits } (x:xs) &= [] : \text{map } (x:) (\text{inits } xs) \end{aligned}$$

(exercise: *inits* matches pattern for *foldr*)

5.5 Efficient scan

- inefficient, as quadratically many applications of *op*
- straightforward to synthesize more efficient implementation

$$\begin{aligned} \text{scanl } op \ e \ [] &= [e] \\ \text{scanl } op \ e \ (x:xs) &= e : \text{scanl } op \ (op \ e \ x) \ xs \end{aligned}$$

- dually,

$$\begin{aligned} \text{scanr} :: (a \rightarrow b \rightarrow b) &\rightarrow b \rightarrow [a] \rightarrow [b] \\ \text{scanr } op \ e &= \text{map } (\text{foldr } op \ e) \circ \text{tails} \end{aligned}$$

$$\begin{aligned} \text{tails} :: [a] &\rightarrow [[a]] \\ \text{tails } [] &= [[]] \\ \text{tails } (x:xs) &= (x:xs) : \text{tails } xs \end{aligned}$$

has more efficient implementation

$$\text{scanr } op \ e = \text{foldr } (\lambda x \ ys \rightarrow op \ x \ (\text{head } ys) : ys) \ [e]$$

5.5 Duality: unfold

- so far we have focused on *consumers*
- *producers* are important too
- producers (unfolds) are *dual* to consumers (folds)
- common pattern

$$\begin{aligned} \text{unfoldr} &:: (b \rightarrow \text{Maybe } (a, b)) \rightarrow (b \rightarrow [a]) \\ \text{unfoldr } \text{coalg } x &= \text{case } \text{coalg } x \text{ of} \\ &\quad \text{Nothing} \rightarrow [] \\ &\quad \text{Just } (a, x') \rightarrow a : \text{unfoldr } \text{coalg } x' \end{aligned}$$

- *unfoldr* is *dual* to *fold* (in what way...?)
- relation to OO iterators?

5.5 Examples of unfold

- $[m..n]$ aka *enumFromTo m n*

$$\begin{aligned} \text{enumFromTo} &:: (\text{Num } a, \text{Ord } a) \Rightarrow a \rightarrow a \rightarrow [a] \\ \text{enumFromTo } m \ n &= \text{unfoldr } (\lambda i \rightarrow \text{if } i > n \text{ then Nothing} \\ &\quad \text{else Just } (i, i + 1)) \ m \end{aligned}$$

- *map* can also be expressed as an unfold

$$\begin{aligned} \text{map} &:: (a \rightarrow b) \rightarrow ([a] \rightarrow [b]) \\ \text{map } f &= \text{unfoldr } (\lambda x \rightarrow \text{case } x \text{ of} \\ &\quad [] \rightarrow \text{Nothing} \\ &\quad a : x' \rightarrow \text{Just } (f \ a, x')) \end{aligned}$$

5.5 Sorting

- given

$$\begin{aligned} \text{insertList} &:: (\text{Ord } a) \Rightarrow \text{Maybe } (a, [a]) \rightarrow [a] \\ \text{insertList } \text{Nothing} &= [] \\ \text{insertList } (\text{Just } (x, [])) &= [x] \\ \text{insertList } (\text{Just } (x, y:ys)) & \\ \quad | x \leq y &= x:y:ys \\ \quad | \text{otherwise} &= y:\text{insertList } (\text{Just } (x, ys)) \end{aligned}$$

we have

$$\begin{aligned} \text{insertSort} &:: (\text{Ord } a) \Rightarrow [a] \rightarrow [a] \\ \text{insertSort} &= \text{fold insertList} \end{aligned}$$

- (exercise: write *insertList* itself as an unfold)

- dually, given

```
deleteMin :: (Ord a) => [a] → Maybe (a, [a])  
deleteMin [ ] = Nothing  
deleteMin (x:xs)  
  = case deleteMin xs of  
    Nothing      → Just (x, [ ])  
    Just (y, ys)  
      | x ≤ y      → Just (x, y:ys)  
      | otherwise → Just (y, x:ys)
```

we have

```
selectSort :: (Ord a) => [a] → [a]  
selectSort = unfoldr deleteMin
```

- (exercise: write *deleteMin* itself as a fold)

5.6 Component-oriented and combinator-style programming

- higher-order functions make a good framework for gluing programs together
- *component-oriented programming*: pluggable units of code, software assembly instead of programming
- manifests itself in a functional language as *combinator style* programming, as higher-order functions sometimes called combinators
- eg functional parsers, see Hutton's *Programming in Haskell*
- eg functional graphics, see Hudak's *The Haskell School of Expression*
- eg functional music composition, ditto

5.6 Music

- Hudak's *Haskore* combinators for expressing musical structure
- primitive entities: notes, rests, durations
- transformations (transposition, tempo-scaling)
- combinations (sequential and parallel, looping)
- translation to MIDI
- algorithmic composition

5.6 A datatype for music

data *Music*

```
= Note Pitch Dur [ NoteAttribute ] -- a note (atomic object)
| Rest Dur -- a rest (atomic object)
| Music :+ : Music -- sequential composition
| Music := : Music -- parallel composition
| Tempo (Ratio Int) Music -- scale the tempo
| Trans Int Music -- transposition
| Instr IName Music -- instrument label
| Player PName Music -- player label
| Phrase [ PhraseAttribute ] Music -- phrase attributes
```

deriving (*Show, Eq*)

```
tequila = tequilaIntro :+: tequilaBody :+: tequilaCoda
```

```
tequilaIntro =
```

```
  drumIntro :+:
```

```
  (drums :=: bass) :+:
```

```
  (drums :=: bass :=: guitar) :+:
```

```
  (drums :=: bass :=: guitar :=: brassIntro)
```

```
tequilaBody =
```

```
  cut 16 (repeatM (
```

```
    twice (drums :=: bass :=: guitar) :=: brass))
```

```
tequilaCoda =
```

```
  drumCoda :=: bassCoda :=: guitarCoda :=: brassCoda
```

```
drumIntro = Instr "Drums" (cut 4 (repeatM (  
    p0 qn :+: p0 en1 :+: p0 en2)))
```

```
drums = Instr "Drums" (drumIntro :=: cut 4 (repeatM (  
    (qn :+: p2 en1 :+: p2 en2) :=: p3 hn)))
```

```
drumCoda = Instr "Drums" (cut 2 drums :+:  
    line [  
        chord [p1 qn, p2 qn, p3 qn],  
        chord [p1 qn, p2 qn, p3 qn],  
        chord [p1 qn, p2 qn, p3 qn],  
        chord [p1 qn, p2 qn, p3 qn, p4 (tie qn wn)] ])
```

```
p1 d = perc RideCymbal2 d [Volume 50]
```

```
p2 d = perc AcousticSnare d [Volume 30]
```

```
p3 d = perc LowTom d [Volume 50]
```

```
p4 d = perc SplashCymbal d [Volume 100]
```

```
p0 d = perc PedalHiHat d [Volume 50]
```

```
bass = Instr "Fretless Bass" bassline
```

```
bassline = cut 4 (repeatM (  
    line [g 2 (tie qn en1) [] ,  
          f 3 (tie en2 en1) [] ,  
          c 3 en2 [] ,  
          a 2 qn [] ]))
```

```
bassCoda = Instr "Fretless Bass" (  
    cut 2 bassline :+ :  
    line [g 2 qn [] , g 2 qn [] , f 2 qn [] , g 2 en1 [] ,  
          en2r , wnr ])
```



```

guitar = Instr "Electric Guitar (jazz)" chordSeq
chordSeq = line [
  g qn, g qn, f (tie qn en1), g (tie en2
    en1), g (tie en2 en1), g en2, f en1, f en2, f en1, f en2,
  g qn, g qn, f (tie qn en1), g (tie en2
    en1), f (tie en2 (tie qn en1)), f en2, f en1, f en2]
where g = eChord G; f = eChord F

```

eChord :: *PitchClass* → *Dur* → *Music*

```

eChord key d
  | pc < pcE = Trans (12 + pc - pcE) (chord (eShape d))
  | otherwise = Trans (pc - pcE) (chord (eShape d))
where
  pc = pitchClass key
  pcE = pitchClass E
  eShape dur = [ n o dur [ Volume 30]
    | (n, o) ← [(e, 3), (b, 3), (e, 4)] ]

```

```
brass = Instr "Brass Section" brassRiff
brassRiff = line [
  g qn, g en1, f en2, a en1, f (tie en2 en1), g (tie en2
    en1), d (tie en2 (tie hn en1)), d en2,
  g qn, g en1, f en2, a en1, f (tie en2 en1), g (tie en2
    (tie dh en1)), d en2,
  g qn, g en1, f en2, a en1, f (tie en2 en1), g (tie en2
    en1), d (tie en2 (tie hn en1)), d en2,
  g qn, g en1, f en2, a en1, f (tie en2 en1), d (tie en2
    (tie hn qn)), en1r, d en2]
where
  g d = Note (G,4) d [ ]
  f d = Note (F,4) d [ ]
  a d = Note (A,4) d [ ]
  d d = Note (D,4) d [ ]
```

```
rep :: (Music → Music) → (Music → Music) → Int →  
      Music → Music  
rep f g 0 m = Rest 0  
rep f g n m = m ::= g (rep f g (n − 1) (f m))  
  
run      = rep (Trans 5) (delay tn) 8 (c 4 tn [ ])  
cascade = rep (Trans 4) (delay en) 8 run  
cascades = rep id (delay sn) 2 cascade  
t4      = test (Instr "piano"  
               (cascades ::+ revM cascades))
```

```
type SNote = [ (AbsPitch, Dur) ]  
pat4' :: [ SNote ]  
pat4' = [ [ (3, 0.5) ], [ (4, 0.25) ], [ (0, 0.25) ], [ (6, 1.0) ] ]  
  
data Cluster = Cl SNote [ Cluster ]  
sim :: [ SNote ] → [ Cluster ]  
sim pat = map mkCl pat  
  where mkCl ns = Cl ns (map (mkCl ∘ addmult ns) pat)  
  addmult = zipWith (λ(p, d) (i, s) → (p + i, d * s))  
  
simFringe n pat = fringe n (Cl [ (0, 0) ] (sim pat))  
fringe 0 (Cl note cls) = [ note ]  
fringe n (Cl note cls) = concat (map (fringe (n - 1)) cls)  
  
sim4s n = l1 :=: l2 where  
  l1 = Instr "flute" s  
  l2 = Instr "bass" (Trans (-36) (revM s))  
  s = Trans 60 (Tempo 2 (simToHask (simFringe n pat4')))
```

5.7 Abstraction, abstraction, abstraction

- question: what are the three most important concepts in programming?
- answer: abstraction, abstraction, abstraction!
- higher-order functions (HOFs) allow you to capture control structures, in particular, common patterns of recursion

Part 6

Laziness

6.0 Outline

Evaluation orders

Demand-driven programming

Efficiency

Infinite data structures

6.1 Evaluation orders

- different evaluation orders are possible
- it matters which we choose
- applicative-order evaluation
- normal-order evaluation
- lazy evaluation
- lazy evaluation in Haskell

6.1 Different evaluation orders

- recall different evaluation orders from before:

\Rightarrow	$\text{square } (3 + 4)$	\Rightarrow	$\text{square } (3 + 4)$
	{ defn of + }		{ defn of <i>square</i> }
	$\text{square } 7$		$(3 + 4) * (3 + 4)$
\Rightarrow	{ defn of <i>square</i> }	\Rightarrow	{ defn of + }
	$7 * 7$		$7 * (3 + 4)$
\Rightarrow	{ defn of * }	\Rightarrow	{ defn of + }
	49		$7 * 7$
		\Rightarrow	{ defn of * }
			49

- not two different answers
- but sometimes no answer at all!
- which order to choose?

6.1 Applicative-order evaluation


- to reduce the application $f e$:
 - first reduce e to normal form
 - then expand definition of f and continue reducing
- (recall, an expression is in *normal form* when it cannot be reduced any further)
- simple and obvious
- easy to implement
- may not terminate when other evaluation orders would

6.1 Normal-order evaluation

- to reduce the application $f\ e$:
 - expand definition of f , substituting e
 - reduce result of expansion
- avoids non-termination, if any evaluation order will
- may involve repeating work

6.1 A third way: Lazy evaluation

- like normal-order evaluation, but instead of copying arguments we share them



- terms are directed graphs, not trees; *graph reduction*
- best of both worlds: evaluates argument only when needed, so terminating, but never evaluates argument more than once, so efficient

6.1 Weak head normal form

- an expression is in *weak head normal form* (WHNF) if it is a lambda, or a constructor applied to zero or more arguments
- eg $\lambda n \rightarrow 2 * 3 + n$, $f x: \text{map } f xs, (1 + 2, 1 - 2)$
- partially normalised
- an expression in normal form is in weak head normal form, but converse does not hold

6.1 Lazy evaluation via **let**

- equivalently, expand application to **let** expression, reduce to WHNF, substitute normalised fragments
- eg consider $sqfst (mkpr\ 3\ 0)$, where

$$\begin{aligned} sqfst\ z &= fst\ z * fst\ z \\ mkpr\ x\ y &= (x + y, x\ 'div'\ y) \end{aligned}$$

- we have

$$\begin{aligned} &sqfst (mkpr\ 3\ 0) \\ \Rightarrow &\mathbf{let}\ z = mkpr\ 3\ 0\ \mathbf{in}\ sqfst\ z \\ \Rightarrow &\mathbf{let}\ z = mkpr\ 3\ 0\ \mathbf{in}\ fst\ z * fst\ z \\ \Rightarrow &\mathbf{let}\ x = 3; y = 0; z = (x + y, x\ 'div'\ y)\ \mathbf{in}\ fst\ z * fst\ z \\ \Rightarrow &\dots \end{aligned}$$

6.1 Lazy evaluation in Haskell

- leftmost outermost *redex* (reducible expression) reduced at each stage, eg in $(2 + 3) * (4 + 5)$, leftmost $+$ first
- pattern-matching may trigger reduction of arguments to WHNF

$$\text{head} [1..1000000] = \text{head} (1 : [1 + 1..1000000]) = 1$$

- patterns matched top to bottom, left to right

$$\text{False} \ \&\& \ x = \text{False}$$

$$\text{True} \ \&\& \ x = x$$

- guards may also trigger reduction

$$\begin{aligned} f \ z \mid \text{fst } z > 0 &= \text{fst } z \\ &\mid \text{otherwise} = \text{snd } z \end{aligned}$$

- local definitions not reduced until needed

$$g \ x = (x \neq 0 \ \&\& \ y < 10) \ \textbf{where} \ y = 1 / x$$

6.2 Demand-driven programming

- lazy evaluation has useful implications for program design
- many computations can be thought of as *pipelines*
- expressed with lazy evaluation, intermediate data structures need not exist all at once
- same effect requires major program surgery in most languages

Slogan: lazy evaluation allows new and better means of modularizing programs

- (but that realization does not help so much in other languages)

6.2 A pipeline

```
    foldl (+) 0 (map square [1..100])  
⇒ foldl (+) 0 (map square (1:[2..100]))  
⇒ foldl (+) 0 (1:map square [2..100])  
⇒ foldl (+) 1 (map square [2..100])  
⇒ foldl (+) 1 (map square (2:[3..100]))  
⇒ foldl (+) 1 (4:map square [3..100])  
⇒ foldl (+) 5 (map square [3..100])  
⇒ ...  
⇒ foldl (+) 14 (map square [4..100])  
⇒ ...  
⇒ 338350
```

6.2 Another pipeline

- insertion sort

```
insertSort :: (Ord a) => [a] -> [a]
insertSort []      = []
insertSort (x:xs) = insert x (insertSort xs)

insert :: (Ord a) => a -> [a] -> [a]
insert a []        = [a]
insert a (b:xs)
  | a <= b          = a:b:xs
  | otherwise       = b:insert a xs
```

- minimum

```
minimum :: (Ord a) => [a] -> a
minimum = head . insertSort
```

- complexity?

6.3 Efficiency

- measure time taken by number of reduction steps
- measure space usage by maximum expression size
- *garbage collection* reclaims discarded space

6.3 Simplifications

- time measure is an approximation, because we ignore time to find redexes
- space measure also an approximation (sharing!)
- eg to evaluate and print `[1..1000]` does not take 1000 units of space
- on the other hand, *space leaks* may surprise

`numbers = [1..1000]`

evaluating and printing `numbers` leaves a pointer, prevents garbage collection

- space occupied by script may grow with use

6.3 Accumulating parameters

- improve efficiency by adding an extra parameter
- common use: to remove expensive $++$
- eg naive reverse

$reverse :: [a] \rightarrow [a]$

$reverse [] = []$

$reverse (x:xs) = reverse xs ++ [x]$

- specify $revCat\ xs\ ys = reverse\ xs ++ ys$
- then $reverse\ xs = revCat\ xs\ []$, and

$revCat :: [a] \rightarrow [a] \rightarrow [a]$

$revCat []\ ys = ys$

$revCat (x:xs)\ ys = revCat\ xs\ (x:ys)$

(we've seen this program before!)

- eg flattening a tree

6.3 Tupling

- dually, improve efficiency by adding an extra result
- eg naive Fibonacci

$$\text{fib } 0 = 0$$

$$\text{fib } 1 = 1$$

$$\text{fib } n = \text{fib } (n - 1) + \text{fib } (n - 2)$$

- introduce $\text{fibtwo } n = (\text{fib } n, \text{fib } (n + 1))$
- then $\text{fib} = \text{fst} \circ \text{fibtwo}$, and

$$\text{fibtwo } 0 = (0, 1)$$

$$\text{fibtwo } n = (b, a + b) \textbf{ where } (a, b) = \text{fibtwo } (n - 1)$$

- eg building balanced binary tree from list

6.3 Fusion and deforestation

- *deforestation* is the removal of intermediate data structures
- *fusion* combines a recursive computation with another computation
- often combined, eg to fuse producer and consumer

- eg deforest treesort to get quicksort

data *Tree a* = *Empty* | *Fork (Tree a) a (Tree a)*

grow :: (*Ord a*) \Rightarrow [*a*] \rightarrow *Tree a*

grow [] = *Empty*

grow (*x* : *xs*) = *Fork (grow littles) x (grow bigs)*

where *littles* = [*a* | *a* \leftarrow *xs*, *a* < *x*]

bigs = [*a* | *a* \leftarrow *xs*, *a* \geq *x*]

inorder :: *Tree a* \rightarrow [*a*]

inorder Empty = []

inorder (Fork t a u) = *inorder t* ++ [*a*] ++ *inorder u*

treeSort :: (*Ord a*) \Rightarrow [*a*] \rightarrow [*a*]

treeSort = *inorder* \circ *grow*

- unfold, fold?
- accumulating parameter?

6.3 Strictness

- recall summing a list:

foldl (+) 0 (map square [1..100])
⇒ foldl (+) 1 (map square [2..100])
⇒ foldl (+) 5 (map square [3..100])
⇒ ...

- this is a white lie; additions are not forced yet:

foldl (+) 0 (map square [1..100])
⇒ foldl (+) (0 + square 1) (map square [2..100])
⇒ foldl (+) ((0 + square 1) + square 2) (map square [3..100])
⇒ ...

- linear space usage, unnecessarily
- what to do about it?

- judicious mix of outermost and innermost evaluation to force additions (safe, because $+$ is strict in both arguments)
- seq* $a\ b$ reduces a to WHNF, then returns b

$$\text{strict} :: (a \rightarrow b) \rightarrow (a \rightarrow b)$$

$$\text{strict } f\ a = \text{seq } a\ (f\ a)$$

- same reductions (on strict functions), but in different order

	$\text{succ } (\text{succ } (8 * 5))$		$\text{strict succ } (\text{strict succ } (8 * 5))$
\Rightarrow	$\text{succ } (8 * 5) + 1$	\Rightarrow	$\text{strict succ } (\text{strict succ } 40)$
\Rightarrow	$((8 * 5) + 1) + 1$	\Rightarrow	$\text{strict succ } (40 + 1)$
\Rightarrow	$(40 + 1) + 1$	\Rightarrow	$\text{strict succ } 41$
\Rightarrow	$41 + 1$	\Rightarrow	$41 + 1$
\Rightarrow	42	\Rightarrow	42

- now try $\text{sfoldl } (+) 0\ (\text{map square } [1..100])$, where

$$\text{sfoldl} :: (b \rightarrow a \rightarrow b) \rightarrow b \rightarrow [a] \rightarrow b$$

$$\text{sfoldl } _\ e\ [] = e$$

$$\text{sfoldl } op\ e\ (x : xs) = \text{strict } (\text{sfoldl } op)\ (op\ e\ x)\ xs$$

6.4 Infinite data structures

- demand-driven evaluation means that programs can manipulate *infinite* data structures
- whole structure is not evaluated at once (fortunately)
- because of laziness, finite result can be obtained from (finite prefix of) infinite data structure
- any recursive datatype has infinite elements, but we will consider only lists

6.4 Infinite lists

- $ones = 1 : ones$
- $[n..] = [n, n + 1, n + 2, \dots]$
- $[n, n + k..] = [n, n + k, n + 2 * k, \dots]$
- $repeat\ n = n : repeat\ n$
- $iterate\ f\ x = x : iterate\ f\ (f\ x)$
- $fibs = 0 : 1 : zipWith\ (+)\ fibs\ (tail\ fibs)$

6.4 No magic

- can apply functions to infinite data structures

filter even [1..] = [2,4,6,8...]

- can return finite results

takeWhile (<10) [1..] = [1,2,3,4,5,6,7,8,9]

- note that these do not always behave like infinite sets in maths

filter (<10) [1..] = [1,2,3,4,5,6,7,8,9]

- to interrupt, ctrl-C

6.4 What does it mean?

- essential idea is that infinite data structure is *limit* of series of *approximations*
- eg infinite list

$[1, 2, 3, 4, 5, \dots]$

is limit of series of approximations

\perp

$1 : \perp$

$1 : 2 : \perp$

$1 : 2 : 3 : \perp$

\dots

6.4 Primes

- recall bounded sequences of primes

primes $m = [n \mid n \leftarrow [1..m], \text{divisors } n == [1, n]]$
divisors $n = [d \mid d \leftarrow [1..n], n \text{ 'mod' } d == 0]$

- infinite sequence of primes

primes $= [n \mid n \leftarrow [1..], \text{divisors } n == [1, n]]$

- much more efficient version: *sieve of Eratosthenes*

primes $= \text{sieve } [2..]$ **where**
sieve $(x:xs) = x:\text{sieve } [y \mid y \leftarrow xs, y \text{ 'mod' } x \neq 0]$

(write *sieve* as an *unfold*)

6.4 Pythagorean triples

- obvious definition

```
pyth = [ (a, b, c)  
         | a ← [1..], b ← [1..], c ← [1..],  
         a * a + b * b == c * c]
```

doesn't work — why?

- instead

```
pyth = [ (a, b, c)  
         | c ← [1..], b ← [1.. c - 1], a ← [1.. b - 1],  
         a * a + b * b == c * c]
```

- then

```
pyth = [(3, 4, 5), (6, 8, 10), (5, 12, 13), ...]
```


6.4 What's the point?

- better abstraction: some real-world entities are infinite
- better modularity: separation of concerns, reuse of components
- provides a model of interaction (but now superseded by monads)
- fun!

Part 7

Reasoning and calculating

7.0 Outline

Reasoning about programs

Equational reasoning

Proof by induction

Program synthesis

Calculating programs

7.1 Reasoning about programs

- functional programs are just equations
- lazy semantics means that rules of ordinary algebra (substitution of equals for equals) apply
- given

three $x = 3$

can replace 3 anywhere by *three* x for any suitably-typed expression x (even $x = 1 / 0$)

- simple proofs by *equational reasoning*

7.2 Equational reasoning

- equations as definitions intended for evaluation
- ... but also useful for reasoning: proofs
- better than testing, because exhaustive
- eg given

$$\text{swap } (x, y) = (y, x)$$

we can show that *swapping* twice is the identity:

$$\begin{aligned} & \text{swap } (\text{swap } (a, b)) \\ = & \quad \{ \text{definition of } \text{swap} \} \\ & \text{swap } (b, a) \\ = & \quad \{ \text{definition of } \text{swap} \} \\ & (a, b) \end{aligned}$$

- program text used as proof rules

7.2 Another simple example

- given

$$\text{curry } f \ a \ b = f \ (a, b)$$

$$\text{fst } (a, b) = a$$

$$\text{const } a \ b = a$$

prove that $\text{const} = \text{curry } \text{fst}$:

$$\begin{aligned} & \text{curry } \text{fst } a \ b \\ = & \quad \{ \text{definition of } \text{curry} \} \\ & \text{fst } (a, b) \\ = & \quad \{ \text{definition of } \text{fst} \} \\ & a \\ = & \quad \{ \text{definition of } \text{const} \} \\ & \text{const } a \ b \end{aligned}$$

7.3 Proof by induction

- proofs about recursive definitions typically require *induction*
- in order to show that some property $P(xs)$ holds for every finite list xs , it suffices to show
 - base case: $P([])$
 - inductive step: if $P(xs)$ holds, then so does $P(x:xs)$ for any x
- this is called *structural induction* (cf mathematical and commonplace induction)
- think of climbing a ladder
- induction is valid for the same reason that recursive definitions are valid: every list is either $[]$ or of the form $x:xs$

7.3 Example: map distributes over ◦

- to prove $\text{map } f (\text{map } g \text{ xs}) = \text{map } (f \circ g) \text{ xs}$ for every finite list xs we use induction on lists
- recall definition of map

$$\begin{aligned}\text{map } f [] &= [] \\ \text{map } f (x : xs) &= f x : \text{map } f xs\end{aligned}$$

- base case

$$\begin{aligned}& \text{map } f (\text{map } g []) \\&= \quad \{ \text{definition of } \text{map} \} \\& \quad \text{map } f [] \\&= \quad \{ \text{definition of } \text{map} \} \\& \quad [] \\&= \quad \{ \text{definition of } \text{map} \} \\& \quad \text{map } (f \circ g) []\end{aligned}$$

- inductive step: assume $\text{map } f (\text{map } g \text{ xs}) = \text{map } (f \circ g) \text{ xs}$; then

$$\begin{aligned} & \text{map } f (\text{map } g (x : \text{xs})) \\ = & \quad \{ \text{definition of } \text{map} \} \\ & \text{map } f (g x : \text{map } g \text{ xs}) \\ = & \quad \{ \text{definition of } \text{map} \} \\ & f (g x) : \text{map } f (\text{map } g \text{ xs}) \\ = & \quad \{ \text{inductive hypothesis} \} \\ & f (g x) : \text{map } (f \circ g) \text{ xs} \\ = & \quad \{ \text{definition of } \circ \} \\ & (f \circ g) x : \text{map } (f \circ g) \text{ xs} \\ = & \quad \{ \text{definition of } \text{map} \} \\ & \text{map } (f \circ g) (x : \text{xs}) \end{aligned}$$

7.3 Example: associativity of append

- to prove

$$xs \mathbin{++} (ys \mathbin{++} zs) = (xs \mathbin{++} ys) \mathbin{++} zs$$

for all finite lists xs, ys, zs

- recall definition

$$\begin{aligned} [] &\mathbin{++} ys = ys \\ (x:xs) \mathbin{++} ys &= x:(xs \mathbin{++} ys) \end{aligned}$$

- use induction over xs (why?)
- base case

$$\begin{aligned} &[] \mathbin{++} (ys \mathbin{++} zs) \\ = &\quad \{ \text{definition of } ++ \} \\ &ys \mathbin{++} zs \\ = &\quad \{ \text{definition of } ++ \} \\ &([] \mathbin{++} ys) \mathbin{++} zs \end{aligned}$$

- inductive step: assume $xs ++ (ys ++ zs) = (xs ++ ys) ++ zs$; then

$$\begin{aligned} & (\lambda: xs) ++ (ys ++ zs) \\ = & \quad \{ \text{definition of } ++ \} \\ & \lambda: (xs ++ (ys ++ zs)) \\ = & \quad \{ \text{inductive hypothesis} \} \\ & \lambda: ((xs ++ ys) ++ zs) \\ = & \quad \{ \text{definition of } ++ \} \\ & (\lambda: (xs ++ ys)) ++ zs \\ = & \quad \{ \text{definition of } ++ \} \\ & ((\lambda: xs) ++ ys) ++ zs \end{aligned}$$

7.3 Structural induction on other datatypes

- exactly the same principle applies for any recursive datatype
- eg for naturals

data *Nat* = *Zero* | *Succ Nat*

one has to show

- ▶ base case: $P(\text{Zero})$
- ▶ inductive step: if $P(n)$ holds, then so does $P(\text{Succ } n)$
- eg for binary trees

data *Btree a* = *Tip a* | *Bin (Btree a) (Btree a)*

one has to show

- ▶ base case: $P(\text{Tip } a)$ holds for any a
- ▶ inductive step: if $P(t)$ and $P(u)$ hold, then so does $P(\text{Bin } t \ u)$

7.3 Inductive proofs about infinite structures

- basic structural induction as above works only for finite data structures
- to prove for infinite (and partially defined) data structures too, must also verify second base case $P(\perp)$
- eg $++$ is associative on infinite and partial lists too, because

$$\begin{aligned}
 & \perp ++ (ys ++ zs) \\
 = & \quad \{ ++ \text{ is strict in its first argument} \} \\
 & \perp \\
 = & \quad \{ ++ \text{ is strict in its first argument} \} \\
 & \perp ++ zs \\
 = & \quad \{ ++ \text{ is strict in its first argument} \} \\
 & (\perp ++ ys) ++ zs
 \end{aligned}$$

- but $\text{filter } (\text{const False}) \text{ xs} = []$ holds only for finite xs , because $\text{filter } (\text{const False}) \perp = \perp$

7.3 When induction doesn't apply

- recall

iterate :: $(a \rightarrow a) \rightarrow a \rightarrow [a]$
iterate $f\ x = x$: *iterate* $f\ (f\ x)$

- expect $\text{map } f\ (\text{iterate } f\ x) = \text{iterate } f\ (f\ x)$
- how to prove? no argument over which to induct
- approximation lemma*: to prove $xs = ys$, suffices to show

$\text{approx } n\ xs = \text{approx } n\ ys$

for every n (by induction), where

$\text{approx } (n + 1)\ [] = \perp$
 $\text{approx } (n + 1)\ (x : xs) = x : \text{approx } n\ xs$

7.4 Program synthesis

- derive program from specification
- eg from specification $\text{revCat } xs \text{ } ys = \text{reverse } xs \text{ } ++ \text{ } ys$,
synthesize

$$\begin{array}{ll}
 & \text{revCat } (x : xs) \text{ } ys \\
 \text{revCat } [] \text{ } ys & = \text{reverse } (x : xs) \text{ } ++ \text{ } ys \\
 = \text{reverse } [] \text{ } ++ \text{ } ys & = (\text{reverse } xs \text{ } ++ [x]) \text{ } ++ \text{ } ys \\
 = [] \text{ } ++ \text{ } ys & = \text{reverse } xs \text{ } ++ ([x] \text{ } ++ \text{ } ys) \\
 = ys & = \text{reverse } xs \text{ } ++ (x : ys) \\
 & = \text{revCat } xs \text{ } (x : ys)
 \end{array}$$

- eg given add on Nat , synthesize sub such that

$$(m \text{ 'add' } n) \text{ 'sub' } n = m$$

7.5 Calculating programs

- sometimes the specification is a (not efficient enough) program
- given enough laws, synthesis may then be a linear calculation
- eg maximum segment sum problem

$mss :: [Integer] \rightarrow Integer$
 $mss = maxlist \circ map\ sum \circ segs$

$segs :: [a] \rightarrow [[a]]$
 $segs = concat \circ map\ inits \circ tails$

$sum :: (Num\ a) \Rightarrow [a] \rightarrow a$
 $sum = foldl\ (+)\ 0$

$maxlist :: (Ord\ a) \Rightarrow [a] \rightarrow a$
 $maxlist = foldr1\ max$

7.5 Laws

- polymorphism of *concat*:

$$\text{map } f \circ \text{concat} = \text{concat} \circ \text{map } (\text{map } f)$$

- 'bookkeeping law': for associative *op*,

$$\text{foldr1 } op \circ \text{concat} = \text{foldr1 } op \circ \text{map } (\text{foldr1 } op)$$

- map* distributes over composition:

$$\text{map } (f \circ g) = \text{map } f \circ \text{map } g$$

- Horner's rule: if *g* is associative with unit *e*, and *g* distributes over *f*, then

$$\text{foldr1 } f \circ \text{scanl } g \ e = \text{foldr } h \ e$$

where $h \ x \ y = f \ e \ (g \ x \ y)$

7.5 Calculation

$$\begin{aligned} & mss \\ = & \quad \{ \text{definition of } mss \} \\ & \quad maxlist \circ map\ sum \circ segs \\ = & \quad \{ \text{definition of } segs \} \\ & \quad maxlist \circ map\ sum \circ concat \circ map\ inits \circ tails \\ = & \quad \{ \text{polymorphism of } concat \} \\ & \quad maxlist \circ concat \circ map\ (map\ sum) \circ map\ inits \circ tails \\ = & \quad \{ \text{bookkeeping law} \} \\ & \quad maxlist \circ map\ maxlist \circ map\ (map\ sum) \circ map\ inits \circ tails \\ = & \quad \{ \text{map distributes over composition} \} \\ & \quad maxlist \circ map\ (maxlist \circ map\ sum \circ inits) \circ tails \end{aligned}$$

$$\begin{aligned} & \text{maxlist} \circ \text{map} (\text{maxlist} \circ \text{map sum} \circ \text{inits}) \circ \text{tails} \\ = & \quad \{ \text{definition of scanl} \} \\ & \text{maxlist} \circ \text{map} (\text{maxlist} \circ \text{scanl} (+) 0) \circ \text{tails} \\ = & \quad \{ \text{Horner's rule: let } h \, x \, y = 0 \, 'max' \, (x + y) \} \\ & \text{maxlist} \circ \text{map} (\text{foldr } h \, 0) \circ \text{tails} \\ = & \quad \{ \text{definition of scanr} \} \\ & \text{maxlist} \circ \text{scanr } h \, 0 \end{aligned}$$

Part 8

Monads

8.0 Outline

Separation of Church and state

The monad interface

Do notation

Define your own monad

The monad type class

Summary

8.1 Separation of Church and state

- a pure functional language such as Haskell is *referentially transparent*
- expressions do not have side-effects
- remember: the sole purpose of an expression is to denote a value
- but what about state-changing computations (eg printing to the console or writing to the file system?)
- how to incorporate these into Haskell?

8.1 Gedankenexperiment

- imagine you are a language designer
- how would you incorporate an outputting computation?

putStr :: *String* → ()

- what's the value and what's the effect of

let *x* = *putStr* "ha" **in** [*x*, *x*]

- and of this one?

[*putStr* "ha", *putStr* "ha"]

- if we noticed different effects, then we would no longer be able to replace equals by equals!

8.1 Monadic IO

- *idea*: *putStr "ha"* has *no* effect at all
- introduce a new type of IO computations

putStr :: *String* → *IO* ()

- *IO a* is type of computation that may do IO, then returns an element of type *a*
- *IO a* can be seen as the type of a *todo list*
- todo list vs actually doing something
- recording an IO computation vs executing an IO computation
- *IO* is a *monad* (more later)
- *main* has type *IO* ()
- *only* the todo list that is bound to *main* is executed

8.1 Interpreting strings

- if evaluator evaluates non-monadic type, prints value; otherwise, performs computation
- strings as values get displayed as strings:

```
? "Hello,\nWorld"  
"Hello,\nWorld"
```

- *putStr* turns a string into an outputting computation:

```
? putStr "Hello,\nWorld"  
Hello,  
World
```

8.2 The monad interface

- *IO a* is an abstract datatype of IO computations
- *return* turns a value into an IO computation that has no effect

return :: *a* → *IO a*

- *m* >> *n* first executes *m* and then *n*

(>>) :: *IO a* → *IO b* → *IO b*

- *m* >>= *n* additionally feeds the result of the first computation into the second

(>>=) :: *IO a* → (*a* → *IO b*) → *IO b*

- every monad supports these three operations
- every monad also supports additional effect-specific operations eg

putStr :: *String* → *IO ()*

getLine :: *IO String*

8.2 Example

- a simple interactive program

```
welcome :: IO ()  
welcome  
  = putStr "Please enter your name.\n" >>  
    getLine >=> \s →  
      putStr ("Welcome " ++ s ++ "!\n")
```

- remember: $\lambda s \rightarrow \dots$ is an anonymous function

8.2 IO computations as first-class citizens

- we can freely mix IO computations with, say, lists

```
main :: IO ()  
main = sequence [ print i | i ← [0..9]]
```

- don't forget the list pattern

```
sequence :: [ IO () ] → IO ()  
sequence [ ]      = return ()  
sequence (a : as) = a >> sequence as
```

(the predefined version of *sequence* is more general)

- IO computations are first-class citizens!
- Haskell is the world's finest imperative language!

8.2 More IO operations

```
print    :: (Show a) ⇒ a → IO ()  
readLn :: (Read a) ⇒ IO a
```

```
putChar :: Char → IO ()  
getChar :: IO Char
```

```
type FilePath = String  
writeFile :: FilePath → String → IO ()  
readFile  :: FilePath → IO String
```

```
data StdGen = ...           -- standard random generator  
class Random where ...    -- randomly generatable  
randomR :: (Random a) ⇒ (a, a) → StdGen → (a, StdGen)  
getStdRandom :: (StdGen → (a, StdGen)) → IO a
```

and many more ...

8.3 Do notation

Special syntactic sugar for monadic expressions.
 Inspired by (in fact, a generalization of) list comprehensions.

$$\begin{aligned}
 \mathbf{do} \{ m \} &= m \\
 \mathbf{do} \{ x \leftarrow m; ms \} &= m \gg= \lambda x \rightarrow \mathbf{do} \{ ms \} \\
 \mathbf{do} \{ m; ms \} &= m \gg= \lambda _ \rightarrow \mathbf{do} \{ ms \} \\
 \mathbf{do} \{ \mathbf{let} \, ds; ms \} &= \mathbf{let} \, ds \mathbf{in} \, \mathbf{do} \{ ms \}
 \end{aligned}$$

where x can appear free in ms .

$$x \leftarrow m$$

Pronounce “ x is drawn from m ”. Note that m has type $IO \, a$,
 whereas x has type a .

8.3 Examples: character I/O

```
putStr, putStrLn :: String → IO ()  
putStr "" = do { return () }  
putStr (c : s) = do { putChar c; putStr s }  
putStrLn s = do { putStr s; putChar '\n' }  
  
getLine' :: IO String  
getLine' = do c ← getChar  
          if c == '\n' then return ""  
          else do s ← getLine'  
          return (c : s)
```

8.3 File I/O

```
processFile :: FilePath → (String → String) → FilePath → IO ()  
processFile inFile f outFile  
  = do s ← readFile inFile  
      let s' = f s  
      writeFile outFile s'
```


8.3 Random numbers

```
import System.Random
```

```
rollDice :: IO Int
```

```
rollDice = getStdRandom (randomR (1,6))
```

```
rollThrice :: IO Int
```

```
rollThrice = do x ← rollDice  
               y ← rollDice  
               z ← rollDice  
               return (x + y + z)
```

8.4 Define your own monad

- *IO* is a *monad*
- monads form *an abstract datatype of computations*.
- computations in general may have *effects*: I/O, exceptions, mutable state, etc.
- monads are a mechanism for cleanly incorporating such impure features in a pure setting
- other monads encapsulate exceptions, state, non-determinism, etc
- the following slides motivate the need for a general notion of computation

8.4 An evaluator

Here's a simple datatype of terms:

```
data Expr = Lit Integer | Div Expr Expr  
  deriving (Show)
```

```
good, bad :: Expr
```

```
good = Div (Lit 7) (Div (Lit 4) (Lit 2))
```

```
bad  = Div (Lit 7) (Div (Lit 2) (Lit 4))
```

... and an evaluation function:

```
eval :: Expr → Integer
```

```
eval (Lit n)    = n
```

```
eval (Div d e) = eval d 'div' eval e
```

8.4 Exceptions

Evaluation may fail, because of division by zero.
Let's handle the exceptional behaviour:

```
data Exc a = Raise Exception | Result a
type Exception = String
```

```
evalE :: Expr → Exc Integer
```

```
evalE (Lit n)    = Result n
```

```
evalE (Div d e) =
```

```
  case evalE d of
```

```
    Raise x  → Raise x
```

```
    Result m → case evalE e of
```

```
      Raise x  → Raise x
```

```
      Result n →
```

```
        if n == 0 then Raise "division by zero"
```

```
        else Result (m `div` n)
```

8.4 Counting

We could instrument the evaluator to count evaluation steps:

newtype Counter a = C (State → (a, State))

type State = Int

run :: Counter a → State → (a, State)

run (C f) = f

evalC :: Expr → Counter Integer

evalC (Lit n) = C (λi → (n, i + 1))

evalC (Div d e) = C (λi →

let (m, i') = run (evalC d) (i + 1)

(n, i'') = run (evalC e) i'

in (m 'div' n, i''))

8.4 Tracing

...or to trace the evaluation steps:

```
newtype Trace a = T (Output, a)
type Output = String
```

```
evalT :: Expr → Trace Integer
evalT (Lit n)    = T (line (Lit n) n, n)
evalT (Div d e) = let
    T (s, m) = evalT d
    T (s', n) = evalT e
    p = m 'div' n
in T (s ++ s' ++ line (Div d e) p, p)
```

```
line :: Expr → Integer → Output
line t n = "    " ++ show t ++ " yields " ++ show n ++ "\n"
```

8.4 Ugly!

- none of these extensions is difficult
- but each is rather awkward, and obscures the previously clear structure
- how can we simplify the presentation?
- what do they have in common?

8.5 The monad type class

These are the methods of a type class:

```
class Monad m where  
  return :: a → m a  
  (>>)    :: m a → m b → m b  
  (>>=)   :: m a → (a → m b) → m b  
  m >> n = m >>= λ_ → n
```

We can also use **do**-notation for *Monad* instances.

8.5 Original evaluator, monadically

$evalM :: (Monad\ m) \Rightarrow Expr \rightarrow m\ Integer$
 $evalM\ (Lit\ n) = return\ n$
 $evalM\ (Div\ d\ e) = evalM\ d \gg= \lambda m \rightarrow$
 $evalM\ e \gg= \lambda n \rightarrow$
 $return\ (m\ 'div'\ n)$

Still pure, but written in the monadic style; much easier to extend.

8.5 Original evaluator, using **do** notation

```
evalM :: (Monad m) ⇒ Expr → m Integer  
evalM (Lit n)    = do return n  
evalM (Div d e) = do m ← evalM d  
                   n ← evalM e  
                   return (m 'div' n)
```

8.5 The exception instance

Exceptions instantiate the class:

```
data Exc a = Raise Exception | Result a
```

```
instance Monad Exc where
```

```
  return a      = Result a
```

```
  Raise e >>= _ = Raise e
```

```
  Result a >>= f = f a
```

The effect-specific behaviour is to throw an exception:

```
throw :: Exception → Exc e
```

```
throw e = Raise e
```

8.5 Exceptional evaluator, monadically

evalE :: *Expr* → *Exc Integer*

evalE (*Lit n*) = **do** return *n*

evalE (*Div d e*) = **do** *m* ← *evalE d*

n ← *evalE e*

if *n* == 0 **then** throw "division by zero"

else return (*m* 'div' *n*)

8.5 The counter instance

Counters instantiate the class:

```
newtype Counter a = C (State → (a, State))
```

```
instance Monad Counter where
```

```
    return a = C (λn → (a, n))
```

```
    ma ≫= f = C (λn → let (a, n') = run ma n in run (f a) n')
```

The effect-specific behaviour is to increment the count:

```
tick :: Counter ()
```

```
tick = C (λn → ((), n + 1))
```

8.5 Counting evaluator, monadically

```
evalC :: Expr → Counter Integer  
evalC (Lit n)    = do tick  
                  return n  
evalC (Div d e) = do tick  
                  m ← evalC d  
                  n  ← evalC e  
                  return (m 'div' n)
```

8.5 The tracing instance

Tracers instantiate the class:

```
newtype Trace a = T (Output, a)
```

```
instance Monad Trace where
```

```
  return a      = T (" ", a)
```

```
  T (s, a)  $\gg=$  f = let T (s', b) = f a in T (s ++ s', b)
```

The effect-specific behaviour is to log some output:

```
trace :: String → Trace ()
```

```
trace s = T (s, ())
```

8.5 Tracing evaluator, monadically

```
evalT :: Expr → Trace Integer  
evalT (Lit n)    = do trace (line (Lit n) n)  
                  return n  
evalT (Div d e) = do m ← evalT d  
                  n  ← evalT e  
                  let p = m 'div' n  
                  trace (line (Div d e) p)  
                  return p
```


8.5 The IO monad

- There's no magic to monads in general: all the monads above are just plain (perhaps higher-order) data, implementing a particular interface.
- But there is one magic monad: the *IO* monad. Its implementation is abstract, hard-wired in the language implementation.

```
data IO a = ...  
instance Monad IO where ...
```

8.6 Abstraction, abstraction, abstraction

- question: what are the three most important concepts in programming?
- answer: abstraction, abstraction, abstraction!
- monads allow you to abstract over patterns of computations (effects)
- Haskell allows you to implement your own computational effect or combination of effects (how cool is this?)
- IO computations are first-class values!
- in general, try to minimize the IO part of your program

END

I hope you've enjoyed the journey!